学習院大学理学部数学科

ステップ･バイ･ステップ式

はじめての Maple

応用編

数学講話 1

▼やっぱり微積分

- 数列と和
- 極限
- 微分法
  - 連続微分
- 級数展開
- 積分法
- 積分の応用
  - 線の積分
  - 回転面の面積
  - ２重積分の計算

▼ いわゆる線形代数

- 行列の演算
  - 行列の定義
  - 行列の演算
  - 行列の操作
- ベクトル
  - ベクトルの定義
  - ベクトルの操作
  - 足し算
  - ベクトルの角度
  - 内積（スカラー積）と外積（ベクトル積）
  - ノルム（ベクトルの大きさ）
- 固有値と固有ベクトル
  - 固有値の計算
  - 固有ベクトルの計算
- １次変換（線形写像）のプロット
- 連立１次方程式の解法

▼ 微分方程式に挑む！

数学モデルの作り方

常微分方程式の定義と解析解

少しだけ常微分方程式のこと

１階常微分方程式

（例１） $\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) = \left(1 - y(x)^2\right) \tan(x)$

（例２） $x\left(\frac{\mathrm{d}}{\mathrm{d}x}\, y(x)\right) = y(x) + \sqrt{x^2 + y(x)^2}$

（例３） $\frac{\mathrm{d}}{\mathrm{d}x}\, y(x) + 2\, y(x) \tan(x) = \sin(x)$

２階常微分方程式の解

１自由度粘性減衰振動系の解

指数関数の特性

１） 異なる２つの実根を持つ場合

a) ともに正の場合

b) 正と負の場合

c) ともに負の場合

2） 異なる２つの虚根を持つ場合

d) 実部が正の場合

e) 実部が負の場合

少しだけ考察

２階常微分方程式の行列表現

連立微分方程式の解法

初期値問題

ラプラス変換による初期値問題の解法

２階常微分方程式の数値解法

▼プログラミング前夜

処理の再利用

プロシージャ化（処理を１行で記述可能な場合）

プロシージャ化（処理を複数行で記述する場合）

グローバル変数とローカル変数

（例）　グローバル変数とローカル変数の違い

引数の型・変数の型

制御構文

条件分岐（if-then-else-end if）

数字の大小を判別する条件分岐

条件分岐を含むプロシージャの作成手順

（参考）引数を比較するプロシージャに変更

繰り返し（for-do-end do）

足し算の実行

繰り返し文の前に

繰り返し文の利用

プロシージャへの拡張

（参考）sum コマンドを用いて実行します

（参考）２つのインデックス操作

それは振動しますか？

Maple による２階常微分方程式の解法と解のプロット

「それは振動しますか？」プロシージャの作成手順

２階常微分方程式の解法をプロシージャ化

プロシージャのパラメータ化

固有値の計算

固有値の実部・虚部抽出

実部の符号と条件分岐

虚部の符号と条件分岐

固有値のプロット

ステップ･バイ･ステップ式 **はじめての Maple** 応用編

やっぱり微積分

微積分に関する操作手順を習得します.

目次

数列と和

初期化します.

```
> restart;
```

第n項が与えられている数列

```
> srn := (1-n^3)/(3*n+1);
```

$$srn := \frac{1-n^3}{3n+1} \tag{1.1}$$

数列の作成（seq コマンドの利用）

```
> seq(srn, n=0..10);
```

$$1,\ 0,\ -1,\ -\frac{13}{5},\ -\frac{63}{13},\ -\frac{31}{4},\ -\frac{215}{19},\ -\frac{171}{11},\ -\frac{511}{25},\ -26,\ -\frac{999}{31} \tag{1.2}$$

n=0..10 の和

```
> sum(srn, 'n'=0..10);
```

$$-\frac{1016815333}{8422700} \tag{1.3}$$

（有限範囲での）プロット

```
> plot(srn, n=0..10);
```



（参考）
プロットの変更や操作は，以下の操作パネルから行うことができます．
操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．

第n項の再表示

```
> srn;
```

$$\frac{1-n^3}{3n+1} \tag{1.4}$$

＋無限大∞までの和

```
> sum(srn, 'n'=0..+infinity);
```

$$-\infty \tag{1.5}$$

無限プロット

```
> plot(srn, n=0..+infinity);
```



第n項の再表示

```
> srn;
```

$$\frac{1-n^3}{3n+1} \qquad \textbf{(1.6)}$$

（参考）極限（limit コマンド）

```
> limit(srn, n=infinity);
```

$$-\infty \qquad \textbf{(1.7)}$$

極限

初期化します.

```
> restart;
```

関数 $\frac{1}{x-1}$ の極限を調べます.

関数 $\frac{1}{x-1}$ の定義

```
> fn := 1/(x-1);
```

$$fn := \frac{1}{x-1} \quad \textbf{(2.1)}$$

$x = 1$ の極限

```
> limit(fn, x=1);
```

$$undefined \quad \textbf{(2.2)}$$

極限値が定まりません.

関数 $\frac{1}{x-1}$ をプロットします.

```
> plot(fn, x=0..2);
```



左側（left）から $x=1$ のときの極限を調べます.

```
> limit(fn, x=1, left);
```

$$-\infty \tag{2.3}$$

左側（right）から $x=1$ のときの極限を調べます.

```
> limit(fn, x=1, right);
```

$$\infty \tag{2.4}$$

関数 $\frac{1}{x-1}$ を再度プロットします.

```
> plot(fn, x=0..5);
```



範囲を拡大します.

```
> plot(fn, x=0..5, y=-5..5);
```


y
x
4
2
0
-2
-4
1
2
3
4
5

微分法

初期化します.

```
> restart;
```

xの関数として f を定義します.

```
> f := sin(a*x)/cos(a*x);
```

$$f := \frac{\sin(a\,x)}{\cos(a\,x)} \tag{3.1}$$

関数 f を x で **1 階微分**します.

```
> df := diff( f, x );
```

$$df := a + \frac{\sin(a\,x)^2\,a}{\cos(a\,x)^2} \tag{3.2}$$

結果を簡単な形に整理します.

```
> simplify(df);
```

$$\frac{a}{\cos(a\,x)^2} \tag{3.3}$$

三角関数 $\sin(a\,x)$ を定義し，xで **2 階微分**します.

```
> f2 := sin(a*x);
```

$$f2 := \sin(a\,x) \tag{3.4}$$

```
> diff(f2, x, x);
```

$$-\sin(a\,x)\,a^2 \tag{3.5}$$

指定方法を一般化します（$の利用）.

```
> diff(f2, x$2);
```

$$-\sin(a\,x)\,a^2 \tag{3.6}$$

x の **n 階微分**になります.

```
> diff(f2, x$n);
```

$$\sin\left(a\,x + \frac{n\,\pi}{2}\right)\,a^n \tag{3.7}$$

2 変数関数 $(x+y)^4$ の微分を確認します.

```
> p := (x + y)^4;
```

$$p := (x+y)^4 \quad \textbf{(3.8)}$$

プロットの応用コマンド（拡張コマンド）をロードします.

```
> with(plots):
```

陰関数（implicit function）をプロットするコマンドを使用して，プロットします.

（参考１）陰関数（implicit function）の例：円の方程式

$x^2+y^2=1$

（参考２）陽関数（explicit function）の例：円の方程式を陽な形で表現

$y=\pm\sqrt{1-x^2}$

$x^2+y^2=1$ をプロットしてみます.

```
> implicitplot( x^2+y^2 = 1, x=-1..1, y=-1..1 );
```



$y = \pm\sqrt{1-x^2}$ をプロットしてみます.

はじめに，式を定義します.

```
> peq :=  sqrt(1-x^2);
  neq := -sqrt(1-x^2);
```

$$peq := \sqrt{1-x^2}$$

$$neq := -\sqrt{1-x^2} \qquad \textbf{(3.9)}$$

それぞれの式をプロットします．オプション scaling=constrained は縦横被を１：１に設定します.

```
> plot(peq, x=-1..1, scaling=constrained);
```




```
> plot(neq, x=-1..1, scaling=constrained);
```



2式を同時にプロットします.

```
> plot([peq, neq], x=-1..1, scaling=constrained);
```



次に 2 変数関数 $(x+y)^4$を implicitplot コマンドでプロットします.

```
> implicitplot(p, x=-5..5, y=-5..5);
```



2変数関数 $(x+y)^4$を展開します.

```
> p := expand(p);
```

$$p := x^4 + 4x^3y + 6x^2y^2 + 4xy^3 + y^4 \quad \textbf{(3.10)}$$

xで1階微分します.

```
> diff(p, x);
```

$$4x^3 + 12x^2y + 12xy^2 + 4y^3 \quad \textbf{(3.11)}$$

y で1階微分します.

```
> diff(p, y);
```

$$4x^3 + 12x^2y + 12xy^2 + 4y^3 \quad \textbf{(3.12)}$$

xで1階微分したものをyでもう一度1階微分します.

```
> diff(p, x, y);
```

$$12x^2 + 24xy + 12y^2 \quad \textbf{(3.13)}$$

（参考）xの２階微分.

```
> diff(p, x, x);
```

$$12\,x^2 + 24\,x\,y + 12\,y^2 \qquad \textbf{(3.14)}$$

三角関数 $\tan(x\,y)$ を定義します.

```
> t := tan(x*y);
```

$$t := \tan(x\,y) \qquad \textbf{(3.15)}$$

$\tan(x\,y)$ を implicitplot コマンドでプロットします.

```
> implicitplot(t, x=-10..10, y=-10..10);
```



連続微分

関数　$\tan(x\,y)$を x で１階微分します．結果を dt に割り当てます.

```
> dt := diff(t, x);
```

$$dt := \left(1 + \tan(x\,y)^2\right) y \qquad \textbf{(3.1.1)}$$

次に，y で 1 階微分したものを ddt に割り当てます．

```
> ddt := diff(dt, y);
```

$$ddt := 2\tan(x\,y)\left(1+\tan(x\,y)^2\right)x\,y+1+\tan(x\,y)^2 \qquad \textbf{(3.1.2)}$$

連続微分（ $\tan(x\,y)$を x ， y で順番に微分）の実行になります．

```
> diff(t, x, y);
```

$$2\tan(x\,y)\left(1+\tan(x\,y)^2\right)x\,y+1+\tan(x\,y)^2 \qquad \textbf{(3.1.3)}$$

級数展開

初期化します．

```
> restart;
```

級数は，数列の無限和になり，数列の和は１変数（x など）の多項式になります．級数展開では，ある無限回微分可能な関数（例えば，$e^{a\,x}$, $\sin(a\,x)$, $\cos(a\,x)$ など）が対象になります．足される項の数が増えると，すなわち多項式の次数が上がると，級数展開される前の関数に近づいていきます（これを，近似と呼びます．正確には，関数の多項式近似になります）．近似は，形や値がよく似ているだけで，必ず真の形や値との間に差が生じます（これを，誤差と呼びます）．ただし，関数を多項式に近似することで，その後の解析がとても容易になります．

関数 sin(x) を，x=0 で（多項式の）次数を 5 として級数展開します．

```
> ser1 := sin(x) = series( sin(x), x=0, 5 );
```

$$ser1 := \sin(x) = x - \frac{1}{6}\,x^3 + \mathrm{O}(x^5) \qquad \textbf{(4.1)}$$

O(x^5) を剰余項と呼びます．多項式近似において，誤差の項を意味します．

（参考） $\mathrm{O}(x^5) = \sin(x) - \left(x - \frac{1}{6}x^3\right)$

展開前の関数と展開後の級数をプロットします．lhs コマンドは式の左辺を抽出し，rhs コマンドは式の右辺を抽出します．

```
> lhs(ser1);
  rhs(ser1);
```

$$\sin(x)$$

$$x - \frac{1}{6}\,x^3 + \mathrm{O}(x^5) \qquad \textbf{(4.2)}$$

```
> plot([ lhs(ser1), rhs(ser1) ], view = [-Pi..Pi, -1.2..1.2], numpoints=
  300 );
```

Warning, unable to evaluate 1 of the 2 functions to numeric values in the region; see the plotting command's help page to ensure the calling sequence is correct



展開後の級数がプロットされません．これは．剰余項が数値でないためにプロットできないためです．ここで，級数を純粋な多項式に変換します．すなわち，剰余項を打ち切ります．このとき発生する誤差を打切り誤差といいます．

```
> pol1 := convert( rhs(ser1), polynom);
```

$$pol1 := x - \frac{1}{6} x^3 \qquad \textbf{(4.3)}$$

剰余項 $O(x^5)$ がなくなります（打ち切られます）．再度，展開前の関数と展開後，級数から多項式に変換した式をプロットします．

```
> plot( [lhs(ser1), pol1], view = [-Pi..Pi, -1.2..1.2], numpoints=300 );
```



展開点 x=0 近傍で，非常によく似た形を示していますが，展開点から離れるに従って差が大きくなっていきます．再度，次数を上げて，近似の度合を調査します．

sin(x) を，x=0 の点において，24 次で級数展開します．

```
> ser2 := sin(x) = series( sin(x), x=0, 24 );
```

$$ser2 := \sin(x) = x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 - \frac{1}{39916800}x^{11} + \frac{1}{6227020800}x^{13} - \frac{1}{1307674368000}x^{15} + \frac{1}{355687428096000}x^{17} - \frac{1}{121645100408832000}x^{19} + \frac{1}{5109094217170944000}x^{21} - \frac{1}{258520167388849766400000}x^{23} + O(x^{24}) \quad \textbf{(4.4)}$$

展開後の級数を多項式に変換します．

```
> pol2 := convert( rhs(ser2), polynom);
```

(4.5)

$$pol2 := x - \frac{1}{6}x^3 + \frac{1}{120}x^5 - \frac{1}{5040}x^7 + \frac{1}{362880}x^9 - \frac{1}{39916800}x^{11} + \frac{1}{6227020800}x^{13} - \frac{1}{1307674368000}x^{15} + \frac{1}{355687428096000}x^{17} - \frac{1}{121645100408832000}x^{19} + \frac{1}{5109094217170944000}x^{21} - \frac{1}{258520167388849766400000}x^{23} \quad \textbf{(4.5)}$$

展開前の関数と展開後の多項式に変換した式を同時にプロットします.

```
> plot( [lhs(ser2), pol2], view = [-Pi..Pi, -1.2..1.2], numpoints=300 );
```



ほぼ重なります. プロットの範囲を広げます.

```
> plot( [lhs(ser2), pol2], view = [-5*Pi..5*Pi, -1.2..1.2], numpoints=300
  );
```



やはり，プロットの両端で差が見られます．（打ち切り）誤差をプロットします．

関数（ sin(x) ）と多項式（ pol2 ）の差を取ります．

```
> err2 := sin(x) - pol2;
```

$$err2 := \sin(x) - x + \frac{x^3}{6} - \frac{x^5}{120} + \frac{x^7}{5040} - \frac{x^9}{362880} + \frac{x^{11}}{39916800} - \frac{x^{13}}{6227020800} + \frac{x^{15}}{1307674368000} - \frac{x^{17}}{355687428096000} + \frac{x^{19}}{121645100408832000} - \frac{x^{21}}{51090942171709440000} + \frac{x^{23}}{25852016738884976640000} \quad \textbf{(4.6)}$$

```
> plot( err2 );
```



次に，関数$\sin(x)+\cos(x)$ を，展開点を x=0 として 6 次で級数展開します．

```
> ser3 := sin(x) + cos(x) = series( sin(x) + cos(x), x=0, 6);
```

$$ser3 := \sin(x)+\cos(x) = 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 + O(x^6) \qquad \textbf{(4.7)}$$

多項式に変換します．

```
> pol3 := convert( rhs(ser3), polynom);
```

$$pol3 := 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 \qquad \textbf{(4.8)}$$

```
> plot( [lhs(ser3), pol3], view = [-Pi..Pi, -2..2], numpoints=300 );
```



以下のように入れ子構造（ネスト構造）で処理を１行にまとめます．

```
> pol3 := convert( rhs(ser3),                          polynom);
> pol3 := convert( sin(x) + cos(x),                    polynom);
> pol3 := convert( series( sin(x) + cos(x), x=0, 6), polynom);
```

最終的に，コマンドのネスト構造が作成されます．

```
> pol3 := convert( series( sin(x) + cos(x), x=0, 6), polynom);
```

$$pol3 := 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 \qquad \textbf{(4.9)}$$

一度で級数展開から多項式への変換が実行されます．さらに，以下のようにパラメータ化します．

関数

```
> f := sin(x) + cos(x);
```

$$f := \sin(x) + \cos(x) \qquad \textbf{(4.10)}$$

展開点

```
> p := 0;
```

$$p := 0 \tag{4.11}$$

次数

```
> k := 6;
```

$$k := 6 \tag{4.12}$$

```
> pol3 := convert( series( f, x=p, k), polynom);
```

$$pol3 := 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 \tag{4.13}$$

いったん，以下のような処理にまとめます．

```
> f := sin(x) + cos(x);
  p := 0;
  k := 6;
  pol3 := convert( series( f, x=p, k), polynom);
```

$$f := \sin(x) + \cos(x)$$

$$p := 0$$

$$k := 6$$

$$pol3 := 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5 \tag{4.14}$$

次に，プロットコマンドと組み合わせます．オプション numpoints=300 はプロットのポイント数を指定します（ポイント数が増えると線が滑らかになります）．

```
> plot([lhs(ser3), pol3], view = [-Pi..Pi, -2..2], numpoints=300 );
```



ただし，rhs(ser3) は，f そのものになりますので，以下のように書き換えます．

```
> plot([f, pol3], view = [-Pi..Pi, -2..2], numpoints=300 );
```



ここで，以下のように整理します．

```
> f := sin(x) + cos(x);
  p := 0;
  k := 6;
  pol3 := convert( series( f, x=p, k), polynom);
  plot([f, pol3], view = [-Pi..Pi, -2..2], numpoints=300 );
```

$$f := \sin(x) + \cos(x)$$

$$p := 0$$

$$k := 6$$

$$pol3 := 1 + x - \frac{1}{2}x^2 - \frac{1}{6}x^3 + \frac{1}{24}x^4 + \frac{1}{120}x^5$$



ここで，次数 k を 1 から 3 まで変化させ（k=1..3），展開後の多項式を式列で作成します．

```
> seq3 := seq( convert( series( f, x=p, k), polynom), k=1..3 );
```

$$seq3 := 1, 1 + x, 1 + x - \frac{1}{2} x^2 \qquad \textbf{(4.15)}$$

さらに，1 から 24 まで（k=1..24）の多項式列を作成します（数が多いので，コロン：で入力を閉じます）．

```
> seq24 := seq( convert( series( f, x=p, k), polynom), k=1..24 ):
```

同じ要領で，各次数に対するプロットを作成します．

```
> plt1 := seq( plot([f, convert( series( f, x=p, k), polynom)], view = [-
  Pi..Pi, -2..2], numpoints=300 ), k=1..3 );
```

$$plt1 := PLOT(\ldots), PLOT(\ldots), PLOT(\ldots) \qquad \textbf{(4.16)}$$

```
> plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [-Pi..Pi, -2..2], numpoints=300 ), k=1..3 ) ];
```

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)] \qquad \textbf{(4.17)}$$

（復習）

PLOT(...) は，プロットコマンド（ plot や plot3d ）から作成される描画イメージを記述したデータです．PLOT(...) をプロットする場合，plots パッケージの display コマンドを使用します．また，display コマンドに複数の PLOT(...) を渡す場合リストにする必要があります．[　] で複数の PLOT(...) を括ると，リストになります．

display コマンドを用いてアニメーションを作成します．plots パッケージをロードします．

```
> with(plots):
```

insequence = true を指定すると，

$plt2 := [PLOT(...), PLOT(...), PLOT(...)]$

の順番でプロットイメージをアニメーションに変換します．

```
> display(plt2, insequence = true);
```



（参考）
アニメーションの操作は，以下の操作パネルから行うことができます．
操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．

次数を 6 にします．

```
> K := 6;
```

$$K := 6 \tag{4.18}$$

k=1..K として，プロットイメージを作成します．

```
> plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [-Pi..Pi, -2..2], numpoints=300 ), k=1..K ) ];
```

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)] \tag{4.19}$$

display コマンドでアニメーションを作成します．

```
> display(plt2,insequence=true);
```



以上の処理を整理します.

```
> f := sin(x) + cos(x);
  p := 0;
  k := 6;
  K := 6;
  plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [-Pi..Pi, -2..2], numpoints=300 ), k=1..K ) ];
  display(plt2,insequence=true);
```

$$f := \sin(x) + \cos(x)$$

$$p := 0$$

$$k := 6$$

$$K := 6$$

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)]$$



最後に，x の範囲もパラメータ化します．

```
> f := sin(x) + cos(x);
  p := 0;
  k := 6;
  K := 15;
  xmin := -2*Pi;
  xmax :=  2*Pi;

  plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [xmin..xmax, -2..2], numpoints=300 ), k=1..K ) ];

  display(plt2,insequence=true);
```

$$f := \sin(x) + \cos(x)$$

$$p := 0$$

$$k := 6$$

$$K := 15$$

$$xmin := -2\,\pi$$

$$xmax := 2\,\pi$$

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)]$$



例えば，関数を変更します．

```
> f := sin(x) + cos(x)^2;
  p := 0;
  k := 6;
  K := 24;
  xmin := -2*Pi;
  xmax :=  2*Pi;

  plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [xmin..xmax, -2..2], numpoints=300 ), k=1..K ) ];

  display(plt2,insequence=true);
```

$$f := \sin(x) + \cos(x)^2$$

$$p := 0$$

$$k := 6$$

$$K := 24$$

$$xmin := -2\,\pi$$

$$xmax := 2\,\pi$$

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots),$$
$$PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots),$$
$$PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots),$$

$PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)]$



例えば，次数を 6 次に固定して，展開点だけを x=-6, -5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5, 6 のように変化させます.

```
> f := sin(x) + cos(x);
  p := 0;
  k := 6; # 次数
  Pmin := -6; # 展開点の最小値
  Pmax :=  6; # 展開点の最大値
  xmin := -3*Pi;
  xmax :=  3*Pi;

  plt2 := [ seq( plot([f, convert( series( f, x=p, k), polynom)], view =
  [xmin..xmax, -2..2], numpoints=300 ), p=Pmin..Pmax ) ];

  display(plt2,insequence=true);
```

$$f := \sin(x) + \cos(x)$$

$$p := 0$$

$$k := 6$$

$$Pmin := -6$$

$$Pmax := 6$$

$$xmin := -3\,\pi$$

$$xmax := 3\,\pi$$

$$plt2 := [PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots), PLOT(\ldots)]$$



積分法

初期化します.

```
> restart;
```

$a\,f(x)$ の不定積分（ a は定数）

```
> int(a*f(x), x);
```

$$\int a\,f(x)\,dx \tag{5.1}$$

```
> simplify((5.1));
```

$$a\left(\int f(x)\,dx\right) \tag{5.2}$$

関数の和 $f(x)+g(x)$ の（不定）積分

```
> int( f(x)+g(x), x );
```

$$\int (f(x)+g(x))\,dx \tag{5.3}$$

```
> expand((5.3));
```

$$\int f(x)\,dx+\int g(x)\,dx \tag{5.4}$$

1 の不定積分

通常，$x+C$ のように積分定数（ C ）が記号としてつきますが， Maple の場合，表示されません.

```
> int(1, x);
```

$$x \tag{5.5}$$

x^aの不定積分

```
> int(x^a, x);
```

$$\frac{x^{a+1}}{a+1} \tag{5.6}$$

$\frac{1}{x}$の不定積分

```
> int( 1/x, x);
```

$$\ln(x) \tag{5.7}$$

$\frac{f'(x)}{f(x)}$の不定積分

```
> int( ( diff(f(x),x)/f(x) ), x);
```

(5.8)

$$\ln(f(x)) \tag{5.8}$$

e^{ax}の不定積分

```
> int( exp(a*x), x);
```

$$\frac{e^{ax}}{a} \tag{5.9}$$

$\sin(ax)$, $\cos(ax)$, および $\frac{1}{\cos(x)^2}$ の不定積分

```
> int( sin(a*x), x);
```

$$-\frac{\cos(ax)}{a} \tag{5.10}$$

```
> int( cos(a*x), x);
```

$$\frac{\sin(ax)}{a} \tag{5.11}$$

```
> int( 1/cos(a*x)^2, x );
```

$$\frac{\sin(ax)}{\cos(ax)\,a} \tag{5.12}$$

式 **(5.12)** を tan の形に変換します.

```
> convert( (5.12), tan );
```

$$\frac{\tan(ax)}{a} \tag{5.13}$$

置換積分法の問題例

```
> f1 := sin(x)*cos(x)^3;
```

$$f1 := \sin(x)\cos(x)^3 \tag{5.14}$$

```
> int( f1, x );
```

$$-\frac{\cos(x)^4}{4} \tag{5.15}$$

部分積分法の問題例

```
> f2 := x*sin(x);
```

$$f2 := x\sin(x) \tag{5.16}$$

```
> int( f2, x );
```

$$\sin(x) - x\cos(x) \tag{5.17}$$

有理関数（分子･分母が多項式の関数）の問題例
（通常，部分分数の形に分解して積分を行います．）

```
> f3 := x/(x^2+x-2);
```

(5.18)

$$f3 := \frac{x}{x^2 + x - 2} \tag{5.18}$$

```
> int(f3, x);
```

$$\frac{\ln(x-1)}{3} + \frac{2\ln(x+2)}{3} \tag{5.19}$$

（参考）部分分数に変換後，積分を実行します．

```
> f3a := convert(f3, parfrac);
```

$$f3a := \frac{2}{3(x+2)} + \frac{1}{3(x-1)} \tag{5.20}$$

```
> int( f3a, x );
```

$$\frac{\ln(x-1)}{3} + \frac{2\ln(x+2)}{3} \tag{5.21}$$

定積分

```
> restart;
```

定積分の記述は以下のように，積分範囲を指定します．

```
> int( f(x), x=a..b );
```

$$\int_a^b f(x)\,dx \tag{5.22}$$

1 の定積分になります．

```
> int( 1, x=1..3 );
```

$$2 \tag{5.23}$$

指数関数の定積分になります．

```
> int( exp(x), x=0..2 );
```

$$-1 + e^2 \tag{5.24}$$

積分の応用

線の積分

初期化します.

```
> restart;
```

（曲線の長さ）

関数 $y=f(x)$ は区間 $[a, b]$ で連続とする．このときの曲線の長さ L は

$$L=\int_a^b \sqrt{1+\{f'(x)\}^2}\,dx$$

である.

直線 $y=x\ (0 \le x \le 1)$ の長さ L

```
> y1 := x;
```

$$y1 := x \qquad \textbf{(6.1.1)}$$

y1 を確認します（プロットします）.

```
> plot(y1, x=0..1);
```



$L = \int_a^b \sqrt{1 + \{f'(x)\}^2}\, dx$ に代入し，評価します．

$f'(x)$ に代入する dy1 を計算します．

```
> dy1 := diff(y1, x);
```

$$dy1 := 1 \qquad \textbf{(6.1.2)}$$

$\sqrt{1 + \{f'(x)\}^2}$ を iy1 として 計算します．

```
> iy1 := sqrt( 1 + dy1^2 );
```

$$iy1 := \sqrt{2} \qquad \textbf{(6.1.3)}$$

最後に $L = \int_a^b \sqrt{1 + \{f'(x)\}^2}\, dx$ を計算します．範囲 x=a..b は x=0..1 になります．

```
> L1 := int( iy1, x=0..1 );
```

$$L1 := \sqrt{2} \qquad \textbf{(6.1.4)}$$

次に，曲線 $y = x\sqrt{x}$ $(0 \le x \le 1)$ の長さ L.

```
> y2 := x*sqrt(x);
```

$$y2 := x^{3/2} \tag{6.1.5}$$

y2 をプロットします.

```
> plot(y2, x=0..1);
```



$\sqrt{1 + \{f'(x)\}^2}$ を iy2 として 計算します.

```
> iy2 := sqrt( 1 + ( diff(y2, x) )^2 );
```

$$iy2 := \frac{\sqrt{4 + 9x}}{2} \tag{6.1.6}$$

最後に $L = \int_a^b \sqrt{1 + \{f'(x)\}^2}\, dx$ を計算します．範囲 x=a..b は x=0..1 になります.

```
> L2 := int( iy2, x=0..1 );
```

(6.1.7)

$$L2 := -\frac{8}{27} + \frac{13\sqrt{13}}{27} \quad \textbf{(6.1.7)}$$

式 **(6.1.7)** を簡単化します.

```
> L2 := simplify(L2);
```

$$L2 := -\frac{8}{27} + \frac{13\sqrt{13}}{27} \quad \textbf{(6.1.8)}$$

（参考）定義式のファンクション化（後述）

`->` はアロー演算子になります（詳細は，プログラミング前夜で説明します）.

```
> L := (y, a, b) -> int( sqrt( 1 + ( diff(y, x) )^2 ), x = a .. b );
```

$$L := (y, a, b) \rightarrow \int_a^b \sqrt{1 + \left(\frac{\partial}{\partial x} y\right)^2}\, dx \quad \textbf{(6.1.9)}$$

```
> L(y1, 0, 1);
```

$$\sqrt{2} \quad \textbf{(6.1.10)}$$

```
> L(y2, 0, 1);
```

$$-\frac{8}{27} + \frac{13\sqrt{13}}{27} \quad \textbf{(6.1.11)}$$

式 $L1 := \sqrt{2}$ を数値化します.

```
> evalf( (6.1.4) );
```

$$1.414213562 \quad \textbf{(6.1.12)}$$

式 $L2 := -\frac{8}{27} + \frac{13\sqrt{13}}{27}$ を評価します.

```
> evalf( (6.1.8) );
```

$$1.439709874 \quad \textbf{(6.1.13)}$$

式 $y1 := x$ 及び式 $y2 := x^{3/2}$ を x=0..1 の範囲でプロットします.

```
> plot([y1, y2], x=0..1);
```



回転面の面積

初期化します.

```
> restart;
```

（回転面の面積）

関数 $y = f(x)$ は区間 $[a, b]$ で連続で $f(x) \geq 0$ とする．このとき， $y = f(x)$ を x 軸のまわりに回転してできる回転面の面積 S は,

$$S = 2\,\pi \left(\int_a^b f(x)\,\sqrt{1 + f'(x)^2}\,dx \right)$$

である.

回転面の面積を求めるファンクション S を定義します.

```
> S := (fx, a, b) -> 2*Pi*( int( fx * sqrt( 1 + ( diff( fx, x ) )^2 ),
```

```
x = a .. b) );
```

$$S := (fx, a, b) \to 2\pi \left(\int_a^b fx \sqrt{1 + \left(\frac{\partial}{\partial x} fx \right)^2} \, dx \right) \quad \textbf{(6.2.1)}$$

次の曲線を x 軸のまわりに回転してできる回転面の面積を求めます.

1. $y = x^3 \quad (0 \le x \le 1)$
2. $y = 2\sqrt{x} \quad (0 \le x \le 1)$

（例１）

$y = x^3 \quad (0 \le x \le 1)$

```
> y1 := x^3;
```

$$y1 := x^3 \quad \textbf{(6.2.2)}$$

y1 を x=0..1 の範囲（区間）でプロットします.

```
> plot(y1, x=0..1);
```



ファンクション S を用いて，区間［0， 1］として関数 y1 の回転面の面積を求めます．

結果を S1 に格納します．

```
> S1 := S(y1, 0, 1);
```

$$S1 := 2\pi\left(\frac{5\sqrt{10}}{27} - \frac{1}{54}\right) \tag{6.2.3}$$

式を S1 を簡単化します．

```
> simplify(S1);
```

$$\frac{\pi\left(10\sqrt{10} - 1\right)}{27} \tag{6.2.4}$$

（例２）

$y = 2\sqrt{x}\ \ (0 \le x \le 1)$

```
> y2 := 2*sqrt(x);
```

$$y2 := 2\sqrt{x} \tag{6.2.5}$$

ファンクション S を用いて，区間［0， 1］として関数 y2 の回転面の面積を求めます．

結果を S2 に格納します．

```
> S2 := S(y2, 0, 1);
```

$$S2 := 2\pi\left(-\frac{4}{3} + \frac{8\sqrt{2}}{3}\right) \tag{6.2.6}$$

式を S2 を簡単化します．

```
> simplify(S2);
```

$$\frac{8\pi\left(-1 + 2\sqrt{2}\right)}{3} \tag{6.2.7}$$

半径 r の球の表面積 S

半円 $y = \sqrt{r^2 - x^2}\ (-r \le x \le r)$ を x 軸のまわりに回転させると球になります．

式を y3 として，定義します．

```
> y3 := sqrt(r^2 - x^2);
```

$$y3 := \sqrt{r^2 - x^2} \tag{6.2.8}$$

式 y3 を r=1 としてプロットします．

```
> param := [r=1];
```

$$param := [r = 1] \tag{6.2.9}$$

param 代入後の式を y3a に格納します.

```
> y3a := eval(y3, param);
```

$$y3a := \sqrt{1 - x^2} \tag{6.2.10}$$

y3a を x=-1..1 の範囲でプロットします．オプション scaling=constrained は，プロットの縦横比を 1 ： 1 にします.

```
> plot( y3a, x=-1..1, scaling = constrained );
```



ファンクション S を用いて，区間［-1，1］として関数 y3a の回転面の面積を求めます．結果を S3a に格納します.

```
> S3a := S(y3a, -1, 1);
```

$$S3a := 4\,\pi \tag{6.2.11}$$

（参考） 球の表面積の公式 $4\pi r^2$ に $r = 1$ を代入します.

```
> S3b := 4*Pi*1^2;
```

$$S3b := 4\,\pi \qquad (6.2.12)$$

２重積分の計算

初期化します.

```
> restart;
```

（２重積分）

$z = f(x, y)$ の D：$a \le x \le b, c \le y \le d$ における２重積分は,

$$\iint_D f(x,y)\,dxdy = \int_a^b \int_c^d f(x,y)\ \mathrm{d}y\,\mathrm{d}x = \int_c^d \int_a^b b(x,y)\ \mathrm{d}y\,\mathrm{d}x$$

２重積分 $\iint_D (x^2 - y)\,dxdy$ $(\mathrm{D}: 0 \le x \le 1, 1 \le y \le 2)$の値を求めます.

```
> fxy := x^2-y;
```

$$fxy := x^2 - y \qquad (6.3.1)$$

（参考）二つの変数（ x, y ）を持つ関数は，plot3d コマンドを用いてプロットします.

```
> plot3d(fxy, x=0..1, y=1..2);
```



（参考）ボックス（目盛り）を追加します．

```
> plot3d(fxy, x=0..1, y=1..2, axes=boxed);
```



（参考） 2 重積分の式表示

コマンドの先頭が大文字の場合：イナート形式＝式の作成までを表示になります.

```
> V := Int( Int( fxy, x=0..1), y=1..2);
```

$$V := \int_1^2 \int_0^1 \left(x^2 - y\right) \, dx \, dy \qquad \textbf{(6.3.2)}$$

2 重積分を評価します.

```
> V := int( int( fxy, x=0..1), y=1..2);
```

$$V := -\frac{7}{6} \qquad \textbf{(6.3.3)}$$

ステップ･バイ･ステップ式 **はじめての Maple** 応用編

いわゆる線形代数

線形代数に関する操作手順を習得します.

目次

行列の演算

初期化します.

```
> restart;
```

はじめに線形代数（ LinearAlgebra ）パッケージをロードします.

```
> with(LinearAlgebra);
```

[*&x, Add, Adjoint, BackwardSubstitute, BandMatrix, Basis, BezoutMatrix, BidiagonalForm, BilinearForm, CARE, CharacteristicMatrix, CharacteristicPolynomial, Column, ColumnDimension, ColumnOperation, ColumnSpace, CompanionMatrix, ConditionNumber, ConstantMatrix, ConstantVector, Copy, CreatePermutation, CrossProduct, DARE, DeleteColumn, DeleteRow, Determinant, Diagonal, DiagonalMatrix, Dimension, Dimensions, DotProduct, EigenConditionNumbers, Eigenvalues, Eigenvectors, Equal, ForwardSubstitute, FrobeniusForm, GaussianElimination, GenerateEquations, GenerateMatrix, Generic, GetResultDataType, GetResultShape, GivensRotationMatrix, GramSchmidt, HankelMatrix, HermiteForm, HermitianTranspose, HessenbergForm, HilbertMatrix, HouseholderMatrix, IdentityMatrix, IntersectionBasis, IsDefinite, IsOrthogonal, IsSimilar, IsUnitary, JordanBlockMatrix, JordanForm, KroneckerProduct, LA_Main, LUDecomposition, LeastSquares, LinearSolve, LyapunovSolve, Map, Map2, MatrixAdd, MatrixExponential, MatrixFunction, MatrixInverse, MatrixMatrixMultiply, MatrixNorm, MatrixPower, MatrixScalarMultiply, MatrixVectorMultiply, MinimalPolynomial, Minor, Modular, Multiply, NoUserValue, Norm, Normalize, NullSpace, OuterProductMatrix, Permanent, Pivot, PopovForm, QRDecomposition, RandomMatrix, RandomVector, Rank, RationalCanonicalForm, ReducedRowEchelonForm, Row, RowDimension, RowOperation, RowSpace, ScalarMatrix, ScalarMultiply, ScalarVector, SchurForm, SingularValues, SmithForm, StronglyConnectedBlocks, SubMatrix, SubVector, SumBasis, SylvesterMatrix, SylvesterSolve, ToeplitzMatrix, Trace, Transpose, TridiagonalForm, UnitVector, VandermondeMatrix, VectorAdd, VectorAngle, VectorMatrixMultiply, VectorNorm, VectorScalarMultiply, ZeroMatrix, ZeroVector, Zip*] **(1.1)**

行列の定義

行列（$r \times c$）のサイズのみで，要素が行列に与えられない場合，すべての値は 0 （デフォルト値）で埋め尽くされます.

```
> M22 := Matrix( 2, 2 );
```

$$M22 := \begin{bmatrix} 0 & 0 \\ 0 & 0 \end{bmatrix} \quad \textbf{(1.1.1)}$$

```
> M33 := Matrix( 3, 3 );
```

$$M33 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.1.2)}$$

```
> M23 := Matrix( 2, 3 );
```

$$M23 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.1.3)}$$

（参考）零行列を明示的に定義する場合は， ZeroMtrix コマンドを使用します．

```
> Z33 := ZeroMatrix( 3, 3 );
```

$$Z33 := \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix} \quad \textbf{(1.1.4)}$$

要素（5）を指定した場合，すべての値がその要素で埋め尽くされます．

```
> M33a := Matrix( 3,3, 5 );
```

$$M33a := \begin{bmatrix} 5 & 5 & 5 \\ 5 & 5 & 5 \\ 5 & 5 & 5 \end{bmatrix} \quad \textbf{(1.1.5)}$$

各要素を指定する場合，以下のような定義方法があります．行列の大きさを宣言し，要素をリストとしてコマンドに与えます．

```
> M33b := Matrix( 3,3, [a,b,c,d,e,f,g,h,i]);
```

$$M33b := \begin{bmatrix} a & b & c \\ d & e & f \\ g & h & i \end{bmatrix} \quad \textbf{(1.1.6)}$$

変数（記号）で要素を定義する場合，以下のような定義方法があります．

```
> M33c := Matrix( 3,3, symbol=a );
```

$$M33c := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} \\ a_{2,1} & a_{2,2} & a_{2,3} \\ a_{3,1} & a_{3,2} & a_{3,3} \end{bmatrix} \quad \textbf{(1.1.7)}$$

< > を用いた行列の定義方法があります．

```
> M33d := < <1,2,3> | <4,5,6> | <7,8,9> >;
```

$$M33d := \begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 9 \end{bmatrix} \quad \textbf{(1.1.8)}$$

単位行列の定義は IdentityMatrix を使用します.

```
> E33 := IdentityMatrix( 3 );
```

$$E33 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix} \quad \textbf{(1.1.9)}$$

対角行列の定義は DiagonalMatrix を使用します.

```
> D33 := DiagonalMatrix( [a,b,c] );
```

$$D33 := \begin{bmatrix} a & 0 & 0 \\ 0 & b & 0 \\ 0 & 0 & c \end{bmatrix} \quad \textbf{(1.1.10)}$$

ランダムに行列の要素を定義するコマンド RandomMatrix があります．要素は実行するたびに異なります.

```
> M33e := RandomMatrix( 3,3 );
```

$$M33e := \begin{bmatrix} 27 & 99 & 92 \\ 8 & 29 & -31 \\ 69 & 44 & 67 \end{bmatrix} \quad \textbf{(1.1.11)}$$

多項式や関数などの式を要素として，行列を定義することができます.

```
> M22f := Matrix( 2,2, [ sin(a*x), cos(b*x), exp(c*x), f(x)] );
```

$$M22f := \begin{bmatrix} \sin(a\,x) & \cos(b\,x) \\ e^{cx} & f(x) \end{bmatrix} \quad \textbf{(1.1.12)}$$

（参考）全要素に diff コマンドを適用します（ x について全要素を 2 階微分）.

```
> map(diff, M22f, x);
```

$$\begin{bmatrix} \cos(a\,x)\,a & -\sin(b\,x)\,b \\ c\,e^{cx} & f'(x) \end{bmatrix} \quad \textbf{(1.1.13)}$$

（参考）全要素に int コマンドを適用します（ x について全要素を積分）.

```
> map(int, M22f, x);
```

$$\begin{bmatrix} -\frac{\cos(a\,x)}{a} & \frac{\sin(b\,x)}{b} \\ \frac{e^{c\,x}}{c} & \int f(x)\,dx \end{bmatrix} \quad \textbf{(1.1.14)}$$

行列の演算

行列の定義

2 × 2 の行列 A

```
> A := Matrix( 2,2, symbol=a );
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} \\ a_{2,1} & a_{2,2} \end{bmatrix} \quad \textbf{(1.2.1)}$$

2 × 2 の行列 B

```
> B := Matrix( 2,2, symbol=b );
```

$$B := \begin{bmatrix} b_{1,1} & b_{1,2} \\ b_{2,1} & b_{2,2} \end{bmatrix} \quad \textbf{(1.2.2)}$$

2 × 2 の行列 C

```
> C := Matrix( 2,2, symbol=c );
```

$$C := \begin{bmatrix} c_{1,1} & c_{1,2} \\ c_{2,1} & c_{2,2} \end{bmatrix} \quad \textbf{(1.2.3)}$$

足し算

```
> A+B;
```

$$\begin{bmatrix} a_{1,1}+b_{1,1} & a_{1,2}+b_{1,2} \\ a_{2,1}+b_{2,1} & a_{2,2}+b_{2,2} \end{bmatrix} \quad \textbf{(1.2.4)}$$

コマンドを使用した足し算の例になります.

```
> MatrixAdd(A,B);
```

$$\begin{bmatrix} a_{1,1}+b_{1,1} & a_{1,2}+b_{1,2} \\ a_{2,1}+b_{2,1} & a_{2,2}+b_{2,2} \end{bmatrix} \quad \textbf{(1.2.5)}$$

引き算

```
> A-B;
```

$$\begin{bmatrix} a_{1,1}-b_{1,1} & a_{1,2}-b_{1,2} \\ a_{2,1}-b_{2,1} & a_{2,2}-b_{2,2} \end{bmatrix} \tag{1.2.6}$$

掛け算

（参考）掛け算にはドット（ . ）を使用します.

```
> A.B;
```

$$\begin{bmatrix} a_{1,1}b_{1,1}+a_{1,2}b_{2,1} & a_{1,1}b_{1,2}+a_{1,2}b_{2,2} \\ a_{2,1}b_{1,1}+a_{2,2}b_{2,1} & a_{2,1}b_{1,2}+a_{2,2}b_{2,2} \end{bmatrix} \tag{1.2.7}$$

逆行列の掛け算

$\frac{A}{B}$ を $A\cdot B^{-1}$ として計算します．数学では行列の割り算が定義されていません.

```
> A.B^(-1);
```

$$\left[\left[\frac{a_{1,1}b_{2,2}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}-\frac{a_{1,2}b_{2,1}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}},-\frac{a_{1,1}b_{1,2}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}+\frac{a_{1,2}b_{1,1}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}\right],\left[\frac{a_{2,1}b_{2,2}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}-\frac{a_{2,2}b_{2,1}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}},-\frac{a_{2,1}b_{1,2}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}+\frac{a_{2,2}b_{1,1}}{b_{1,1}b_{2,2}-b_{1,2}b_{2,1}}\right]\right] \tag{1.2.8}$$

逆行列を求める MatrixInverse コマンドもあります.

```
> MatrixInverse(A);
```

$$\begin{bmatrix} \frac{a_{2,2}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} & -\frac{a_{1,2}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} \\ -\frac{a_{2,1}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} & \frac{a_{1,1}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} \end{bmatrix} \tag{1.2.9}$$

A^{-1} も A の逆行列を求めることができます.

```
> A^(-1);
```

$$\begin{bmatrix} \frac{a_{2,2}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} & -\frac{a_{1,2}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} \\ -\frac{a_{2,1}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} & \frac{a_{1,1}}{a_{1,1}a_{2,2}-a_{1,2}a_{2,1}} \end{bmatrix} \quad (1.2.10)$$

行列式は Determinant コマンドを使用します.

```
> Determinant(A);
```

$$a_{1,1}a_{2,2}-a_{1,2}a_{2,1} \quad (1.2.11)$$

行列のランクを求めるには，Rank コマンドを使用します.

```
> Rank(A);
```

$$2 \quad (1.2.12)$$

行列のノルムを計算します．はじめに，３×３の行列 N1 を定義します.

```
> N1 := Matrix( 3,3, [a, b, c, d, e, f, g, h, i] );
```

$$N1 := \begin{bmatrix} a & b & c \\ d & e & f \\ g & h & i \end{bmatrix} \quad (1.2.13)$$

行列のノルムを求めるには，MatrixNorm コマンドを用います.

```
> MatrixNorm( N1, 1 );
```

$$\max(|a|+|d|+|g|, |b|+|e|+|h|, |c|+|f|+|i|) \quad (1.2.14)$$

max は最大値を求めるコマンドになります.

（参考）｜ ｜は絶対値を表します.

（例）

行列のノルムを実際に求めてみます．３×３の行列 N1 を定義します.

```
> N2 := Matrix( 3,3, [1, -2, 3, -4, -5, 6, -7, 8, -9] );
```

$$N2 := \begin{bmatrix} 1 & -2 & 3 \\ -4 & -5 & 6 \\ -7 & 8 & -9 \end{bmatrix} \quad (1.2.15)$$

行列 N2 のノルムを計算します.

```
> MatrixNorm( N2, 1 );
```

$$18 \qquad \textbf{(1.2.16)}$$

（参考）式 $\max(|a|+|d|+|g|, |b|+|e|+|h|, |c|+|f|+|i|)$ に実際の値を代入して確認します．

```
> max( abs( 1) + abs(-4) + abs(-7),
       abs(-2) + abs(-5) + abs( 8),
       abs( 3) + abs( 6) + abs(-9) );
```

$$18 \qquad \textbf{(1.2.17)}$$

行列の操作

初期化します．

```
> restart;
```

はじめに，線形代数パッケージをロードします．

```
> with(LinearAlgebra):
```

（参考）コロン（：）で，結果を非表示にしています．

行列 M の転置

```
> M := Matrix( 6,6, symbol=m );
```

$$M := \begin{bmatrix} m_{1,1} & m_{1,2} & m_{1,3} & m_{1,4} & m_{1,5} & m_{1,6} \\ m_{2,1} & m_{2,2} & m_{2,3} & m_{2,4} & m_{2,5} & m_{2,6} \\ m_{3,1} & m_{3,2} & m_{3,3} & m_{3,4} & m_{3,5} & m_{3,6} \\ m_{4,1} & m_{4,2} & m_{4,3} & m_{4,4} & m_{4,5} & m_{4,6} \\ m_{5,1} & m_{5,2} & m_{5,3} & m_{5,4} & m_{5,5} & m_{5,6} \\ m_{6,1} & m_{6,2} & m_{6,3} & m_{6,4} & m_{6,5} & m_{6,6} \end{bmatrix} \qquad \textbf{(1.3.1)}$$

転置は，Transpose コマンドを使用します．

```
> Transpose(M);
```

(1.3.2)

$$\begin{bmatrix} m_{1,1} & m_{2,1} & m_{3,1} & m_{4,1} & m_{5,1} & m_{6,1} \\ m_{1,2} & m_{2,2} & m_{3,2} & m_{4,2} & m_{5,2} & m_{6,2} \\ m_{1,3} & m_{2,3} & m_{3,3} & m_{4,3} & m_{5,3} & m_{6,3} \\ m_{1,4} & m_{2,4} & m_{3,4} & m_{4,4} & m_{5,4} & m_{6,4} \\ m_{1,5} & m_{2,5} & m_{3,5} & m_{4,5} & m_{5,5} & m_{6,5} \\ m_{1,6} & m_{2,6} & m_{3,6} & m_{4,6} & m_{5,6} & m_{6,6} \end{bmatrix} \quad \textbf{(1.3.2)}$$

$A^{\%T}$ によっても転置を行うことが出来ます.

```
> M^%T;
```

$$\begin{bmatrix} m_{1,1} & m_{2,1} & m_{3,1} & m_{4,1} & m_{5,1} & m_{6,1} \\ m_{1,2} & m_{2,2} & m_{3,2} & m_{4,2} & m_{5,2} & m_{6,2} \\ m_{1,3} & m_{2,3} & m_{3,3} & m_{4,3} & m_{5,3} & m_{6,3} \\ m_{1,4} & m_{2,4} & m_{3,4} & m_{4,4} & m_{5,4} & m_{6,4} \\ m_{1,5} & m_{2,5} & m_{3,5} & m_{4,5} & m_{5,5} & m_{6,5} \\ m_{1,6} & m_{2,6} & m_{3,6} & m_{4,6} & m_{5,6} & m_{6,6} \end{bmatrix} \quad \textbf{(1.3.3)}$$

要素の抽出

```
> M[1,1];
```

$$m_{1,1} \quad \textbf{(1.3.4)}$$

```
> M[2,3];
```

$$m_{2,3} \quad \textbf{(1.3.5)}$$

部分行列

1行目から3行目まで. 1列目から3列目までを取り出します.

```
> M[ 1..3, 1..3 ];
```

$$\begin{bmatrix} m_{1,1} & m_{1,2} & m_{1,3} \\ m_{2,1} & m_{2,2} & m_{2,3} \\ m_{3,1} & m_{3,2} & m_{3,3} \end{bmatrix} \quad \textbf{(1.3.6)}$$

4行目から5行目まで. 3列目から6列目までを取り出します.

```
> M[ 4..5, 3..6 ];
```

$$\begin{bmatrix} m_{4,3} & m_{4,4} & m_{4,5} & m_{4,6} \\ m_{5,3} & m_{5,4} & m_{5,5} & m_{5,6} \end{bmatrix} \quad (1.3.7)$$

1 行目だけを取り出します.

```
> M[ 1, 1..6 ];
```

$$\begin{bmatrix} m_{1,1} & m_{1,2} & m_{1,3} & m_{1,4} & m_{1,5} & m_{1,6} \end{bmatrix} \quad (1.3.8)$$

1 列目だけを取り出します.

```
> M[ 1..6, 1];
```

$$\begin{bmatrix} m_{1,1} \\ m_{2,1} \\ m_{3,1} \\ m_{4,1} \\ m_{5,1} \\ m_{6,1} \end{bmatrix} \quad (1.3.9)$$

最後の要素を -1 とすることで，要素数が不明な行列もm行目だけ，あるいはｎ列目だけを取り出すことができます．以下は，それぞれ行ベクトル，列ベクトルになります.

```
> M[ 1, 1..-1 ];
```

$$\begin{bmatrix} m_{1,1} & m_{1,2} & m_{1,3} & m_{1,4} & m_{1,5} & m_{1,6} \end{bmatrix} \quad (1.3.10)$$

```
> M[ 1..-1, 1];
```

$$\begin{bmatrix} m_{1,1} \\ m_{2,1} \\ m_{3,1} \\ m_{4,1} \\ m_{5,1} \\ m_{6,1} \end{bmatrix} \quad (1.3.11)$$

ベクトル

初期化します.

```
> restart;
```

はじめに，線形代数パッケージをロードします.

```
> with(LinearAlgebra):
```

（参考）コロン（：）で，結果を非表示にしています.

ベクトルの定義

ベクトルのサイズのみで，要素がベクトルに与えられない場合，すべての値は 0 （デフォルト値）で埋め尽くされます．特に指定がなければ，列ベクトルが定義されます.

```
> Vector(2);
```

$$\begin{bmatrix} 0 \\ 0 \end{bmatrix} \tag{2.1.1}$$

（参考）零ベクトルを明示的に定義する場合は， ZeroVector コマンドを使用します.

```
> ZeroVector(3);
```

$$\begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} \tag{2.1.2}$$

ベクトルのサイズを指定し，要素をすべて 5 で定義します.

```
> Vector( 1..3, 5 );
```

$$\begin{bmatrix} 5 \\ 5 \\ 5 \end{bmatrix} \tag{2.1.3}$$

ベクトルの要素を（リストで）指定します.

```
> Vector([1,2,3]);
```

$$\begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \tag{2.1.4}$$

行ベクトルを定義します.

```
> Vector[row]([1,2,3]);
```

$$\begin{bmatrix} 1 & 2 & 3 \end{bmatrix} \tag{2.1.5}$$

記号で要素を指定します.

```
> Vector(3, symbol=v);
```

$$\begin{bmatrix} v_1 \\ v_2 \\ v_3 \end{bmatrix} \tag{2.1.6}$$

山括弧 < > を用いて，ベクトルを定義することもできます．また，カンマ （, ）は要素を縦方向に並べる働きがあり，バーティカルバー （|）は要素を横方向に並べる働きがあります.

```
> vc := <1,2,3>;
```

$$vc := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix} \tag{2.1.7}$$

```
> vr := <1|2|3>;
```

$$vr := \begin{bmatrix} 1 & 2 & 3 \end{bmatrix} \tag{2.1.8}$$

ランダムにベクトルの要素を定義するコマンド RandomVector があります.
要素は実行するたびに異なります.

```
> RandomVector(3);
```

$$\begin{bmatrix} 92 \\ -31 \\ 67 \end{bmatrix} \tag{2.1.9}$$

シーケンスの形で，基底ベクトルを定義します．シーケンスはカンマ （, ） で区切られた Maple オブジェクトのひとつになります.

```
> UnitVector(1,3), UnitVector(2,3), UnitVector(3,3);
```

$$\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 1 \\ 0 \end{bmatrix}, \begin{bmatrix} 0 \\ 0 \\ 1 \end{bmatrix} \tag{2.1.10}$$

ベクトルの操作

列ベクトルを定義します.

```
> v1 := Vector([a, b, c]);
```

$$v1 := \begin{bmatrix} a \\ b \\ c \end{bmatrix} \quad \textbf{(2.2.1)}$$

要素を抽出します.

```
> v1[1]; v1[2]; v1[3];
```

$$a$$
$$b$$
$$c \quad \textbf{(2.2.2)}$$

行ベクトルを定義します.

```
> v2 := Vector[row]([x, y, z]);
```

$$v2 := \begin{bmatrix} x & y & z \end{bmatrix} \quad \textbf{(2.2.3)}$$

要素を抽出します.

```
> v2[1]; v2[2]; v2[3];
```

$$x$$
$$y$$
$$z \quad \textbf{(2.2.4)}$$

足し算

列ベクトル p を定義します.

```
> p := Vector(3, symbol=a);
```

$$p := \begin{bmatrix} a_1 \\ a_2 \\ a_3 \end{bmatrix} \quad \textbf{(2.3.1)}$$

列ベクトル q を定義します.

```
> q := Vector(3, symbol=b);
```

$$q := \begin{bmatrix} b_1 \\ b_2 \\ b_3 \end{bmatrix} \quad \textbf{(2.3.2)}$$

```
> VectorAdd(p, q);
```

(2.3.3)

$$\begin{bmatrix} a_1 + b_1 \\ a_2 + b_2 \\ a_3 + b_3 \end{bmatrix} \quad \textbf{(2.3.3)}$$

ベクトルの角度

直交する２つのベクトルを定義します．

（２つのベクトルがなす角度は $\frac{\pi}{2}$ になります）

```
> v3 := Vector([1,0]);
  v4 := Vector([0,1]);
```

$$v3 := \begin{bmatrix} 1 \\ 0 \end{bmatrix}$$

$$v4 := \begin{bmatrix} 0 \\ 1 \end{bmatrix} \quad \textbf{(2.4.1)}$$

VectorAngle コマンドを用いて，２つのベクトルの角度を計算します．

```
> a := VectorAngle(v3, v4);
```

$$a := \frac{\pi}{2} \quad \textbf{(2.4.2)}$$

内積（スカラー積）と外積（ベクトル積）

初期化します．

```
> restart;
```

ここでは，ベクトルのプロットをするために，学生向けのベクトル解析（VectorCalculus）パッケージをロードします．

```
> with(Student[VectorCalculus]);
```

[*&x*, `*`, `+`, `-`, `.`, *<,>*, *<|>*, *About*, *ArcLength*, *BasisFormat*, *Binormal*, *ConvertVector*, *CrossProduct*, *Curl*, *Curvature*, D, *Del*, *DirectionalDiff*, *Divergence*, *DotProduct*, *FlowLine*, *Flux*, *GetCoordinates*, *GetPVDescription*, *GetRootPoint*, *GetSpace*, *Gradient*, *Hessian*, *IsPositionVector*, *IsRootedVector*, *IsVectorField*, *Jacobian*, *Laplacian*, *LineInt*, *MapToBasis*, ∇, *Norm*, *Normalize*, *PathInt*, *PlotPositionVector*, *PlotVector*, *PositionVector*, *PrincipalNormal*, *RadiusOfCurvature*, *RootedVector*, *ScalarPotential*, *SetCoordinates*, *SpaceCurve*, *SpaceCurveTutor*, *SurfaceInt*, *TNBFrame*, *Tangent*, *TangentLine*, *TangentPlane*, *TangentVector*, *Torsion*, *Vector*, *VectorField*, *VectorFieldTutor*, *VectorPotential*, *VectorSpace*, *diff*, *evalVF*, *int*, *limit*, *series*] **(2.5.1)**

2つの（3次元）ベクトル v1，v2 を定義します．

e_x, e_y, e_z は，それぞれ3次元の直交座標系（デカルト座標系）の単位ベクトルを表します．

（参考） $e_x, e_y, e_z = \left[\begin{matrix}1\\0\\0\end{matrix}\right], \left[\begin{matrix}0\\1\\0\end{matrix}\right], \left[\begin{matrix}0\\0\\1\end{matrix}\right]$

```
> v1 := Vector([1, 2, 1]);
```

$$v1 := e_x + 2e_y + e_z \qquad \textbf{(2.5.2)}$$

```
> v2 := Vector([-3, 1, -2]);
```

$$v2 := -3e_x + e_y - 2e_z \qquad \textbf{(2.5.3)}$$

ベクトルをプロットします．

```
> PlotVector([v1,v2], color=[red,  blue], axes=boxed, scaling=
  constrained );
```



ベクトル v1 と v2 の内積（スカラー積）を DotProduct コマンドを用いて計算します．

```
> dpv := DotProduct(v1, v2);
```

$$dpv := -3 \tag{2.5.4}$$

ベクトル v1 と v2 の外積（ベクトル積）を CrossProduct コマンドを用いて計算します.

```
> cpv := CrossProduct(v1, v2);
```

$$cpv := -5e_x - e_y + 7e_z \tag{2.5.5}$$

cpv をプロットしてみます.

```
> PlotVector(cpv, color=green, axes=boxed, scaling=constrained );
```



ベクトル v1，v2，cpv を同時にプロットします.

```
> PlotVector([v1, v2, cpv], color=[red, blue, green], axes=boxed,
  scaling=constrained );
```



ノルム（ベクトルの大きさ）

初期化します.

```
> restart:
```

線形代数（LinearAlgebra）パッケージをロードします.

```
> with(LinearAlgebra):
```

（参考）コロン（：）で，結果を非表示にしています.

ベクトルを定義します.

```
> v := Vector([a, b]);
```

$$v := \begin{bmatrix} a \\ b \end{bmatrix} \qquad \textbf{(2.6.1)}$$

ベクトル v の 1-ノルム

```
> Norm(v, 1);
```

$$|a| + |b| \tag{2.6.2}$$

ベクトル v の 2-ノルム

```
> Norm(v, 2);
```

$$\sqrt{|a|^2 + |b|^2} \tag{2.6.3}$$

ベクトル v の 3-ノルム

```
> Norm(v, 3);
```

$$\left(|a|^3 + |b|^3\right)^{1/3} \tag{2.6.4}$$

ベクトル v の 3/2-ノルム

```
> Norm(v, 3/2);
```

$$\left(|a|^{3/2} + |b|^{3/2}\right)^{2/3} \tag{2.6.5}$$

ベクトル v の 無限大-ノルム

```
> Norm(v, infinity);
```

$$\max(|a|, |b|) \tag{2.6.6}$$

固有値と固有ベクトル

初期化します.

```
> restart;
```

線形代数（LinearAlgebra）パッケージをロードします.

```
> with(LinearAlgebra):
```

（参考）コロン（：）で，結果を非表示にしています.

固有値の計算

2×2の行列 A を定義し，その固有値を計算します.

```
> A := Matrix( 2,2, [2, 0, 7, -5] );
```

$$A := \begin{bmatrix} 2 & 0 \\ 7 & -5 \end{bmatrix} \tag{3.1.1}$$

行列 A の固有値を Eigenvalues コマンドを用いて求めます.

```
> r := Eigenvalues(A);
```

$$r := \begin{bmatrix} 2 \\ -5 \end{bmatrix} \tag{3.1.2}$$

以下は，教科書にあるような手順で固有値を求めています.

はじめ，2×2の単位行列を定義します.

```
> E := DiagonalMatrix([1,1]);
```

$$E := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \tag{3.1.3}$$

A を正方行列，λ を固有値，x を固有ベクトルとした場合，$A\,x = \lambda\,x$ を変形して得られる以下の式より，固有方程式（特性方程式）を導きます.

$|A - \lambda E| = 0$

```
> M := A - lambda*E;
```

$$M := \begin{bmatrix} 2-\lambda & 0 \\ 7 & -5-\lambda \end{bmatrix} \tag{3.1.4}$$

行列式を求める Determinant コマンドを使用します．

```
> ceq := Determinant(M) = 0;
```

$$ceq := (-2+\lambda)(5+\lambda) = 0 \quad \textbf{(3.1.5)}$$

固有方程式（特性方程式） $ceq := (-2+\lambda)(5+\lambda) = 0$ の解を求めます．

```
> R := solve(ceq, lambda);
```

$$R := 2, -5 \quad \textbf{(3.1.6)}$$

$r := \begin{bmatrix} 2 \\ -5 \end{bmatrix}$ と同じように固有値が導かれます．

（参考）固有値方程式と特性方程式は同じものです．特性方程式として解を求めた場合，導かれた根を特性根と呼びます．

固有ベクトルの計算

はじめに定義した 2 × 2 の行列 A について，その固有値と固有ベクトルを計算します．Eigenvalues コマンドは，固有値と固有ベクトルを同時に計算します．そこで，2 つの変数 e，v を用意して，

```
> ( e, v ) := Eigenvectors(A);
```

$$e, v := \begin{bmatrix} 2 \\ -5 \end{bmatrix}, \begin{bmatrix} 1 & 0 \\ 1 & 1 \end{bmatrix} \quad \textbf{(3.2.1)}$$

2 つの固有値をそれぞれ別々の変数 e1，e2 に割り当てます．

```
> e1 := e[1];
  e2 := e[2];
```

$$e1 := 2$$
$$e2 := -5 \quad \textbf{(3.2.2)}$$

2 つの固有ベクトルをそれぞれ別々の変数 v1，v2 に割り当てます．

```
> v1 := v[1..-1, 1]; # １列目を抽出
  v2 := v[1..-1, 2]; # ２列目を抽出
```

$$v1 := \begin{bmatrix} 1 \\ 1 \end{bmatrix}$$
$$v2 := \begin{bmatrix} 0 \\ 1 \end{bmatrix} \quad \textbf{(3.2.3)}$$

A を正方行列，λ を固有値，x を固有ベクトルとした場合，$A\,x = \lambda\,x$ の関係式が成り立ちますので，求められた固有値と固有ベクトルをそれぞれ関係式に代入してみます．

$A\,v1 = e1\,v1$ より，以下のようになります．

（参考）ドット（．）は内積を表します．

```
> A.v1 = e1.v1;
```

$$\begin{bmatrix} 2 \\ 2 \end{bmatrix} = \begin{bmatrix} 2 \\ 2 \end{bmatrix} \tag{3.2.4}$$

同様に，$A\,v2 = e2\,v2$ を調べます．

```
> A.v2 = e2.v2;
```

$$\begin{bmatrix} 0 \\ -5 \end{bmatrix} = \begin{bmatrix} 0 \\ -5 \end{bmatrix} \tag{3.2.5}$$

1次変換（線形写像）のプロット

初期化します．

```
> restart;
```

1次変換（線形写像）のプロットコマンドを使用するために，学生向け線形代数（LinearAlgebra）パッケージをロードします．

```
> with(Student[LinearAlgebra]):
```

（参考）コロン（：）で，結果を非表示にしています．

表現行列を単位行列として定義します．

```
> E := IdentityMatrix( 2 );
```

$$E := \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \tag{4.1}$$

1次変換を LinearTransformPlot を用いてプロットします．左側の単位円（半径1の円）が変換されます．単位行列の場合，自分自身に変換されます（恒等変換）．

```
> LinearTransformPlot(E);
```

The Image of the Unit Circle
In the Plane



ある表現行列を定義します.

```
> M := Matrix( 2,2, [-2,1,3,-2] );
```

$$M := \begin{bmatrix} -2 & 1 \\ 3 & -2 \end{bmatrix} \quad \textbf{(4.2)}$$

1 次変換を実行します.

```
> LinearTransformPlot(M);
```

The Image of the Unit Circle
In the Plane



表現行列を３次元に拡大します.

```
> A := Matrix( 3,3, [-2,3,3, 2,-3,2, 3,2,-3]);
```

$$A := \begin{bmatrix} -2 & 3 & 3 \\ 2 & -3 & 2 \\ 3 & 2 & -3 \end{bmatrix} \quad \textbf{(4.3)}$$

１次変換を実行します.

```
> LinearTransformPlot(A);
```

The Image of the Unit Sphere
In 3-Space



連立１次方程式の解法

初期化します．

```
> restart;
```

線形代数（LinearAlgebra）パッケージをロードします．

```
> with(LinearAlgebra):
```

（参考）コロン（：）で，結果を非表示にしています．

はじめに連立１次方程式を（係数を文字として）定義します．

```
> eq := [ a11*x+a12*y+a13*z=p, a21*x+a22*y+a23*z=q, a31*x+a32*y+a33*z=r ];
```

$$eq := [a11\,x + a12\,y + a13\,z = p,\ a21\,x + a22\,y + a23\,z = q,\ a31\,x + a32\,y + a33\,z = r] \quad \textbf{(5.1)}$$

各１次方程式を表示します．

```
> eq[1];
  eq[2];
  eq[3];
```

$$a11\,x + a12\,y + a13\,z = p$$
$$a21\,x + a22\,y + a23\,z = q$$
$$a31\,x + a32\,y + a33\,z = r \quad \textbf{(5.2)}$$

係数を定義します．

```
> cffs := [a11=1, a12=1, a13=-1, a21=2, a22=2, a23=0, a31=0, a32=1, a33=3,
  p=5, q=-1, r=2];
```

$$cffs := [a11 = 1,\ a12 = 1,\ a13 = -1,\ a21 = 2,\ a22 = 2,\ a23 = 0,\ a31 = 0,\ a32 = 1,\ a33 = 3,\ p = 5,\ q = -1,\ r = 2] \quad \textbf{(5.3)}$$

係数を，式 **(5.1)** に代入します．

```
> eqs := eval(eq, cffs);
```

$$eqs := [x + y - z = 5,\ 2\,x + 2\,y = -1,\ y + 3\,z = 2] \quad \textbf{(5.4)}$$

従属変数を定義します．

```
v := [x, y, z];
```

$$v := [x, y, z] \quad \textbf{(5.5)}$$

solve コマンドを使用して連立1次方程式を解きます．

```
> sol := solve(eqs, v);
```

$$sol := \left[\left[x = -19, y = \frac{37}{2}, z = -\frac{11}{2}\right]\right] \quad (5.6)$$

以下は行列操作によって連立方程式を解く手順になります.

係数行列を抽出します.

```
> (A, b) := GenerateMatrix( eqs, v );
```

$$A, b := \begin{bmatrix} 1 & 1 & -1 \\ 2 & 2 & 0 \\ 0 & 1 & 3 \end{bmatrix}, \begin{bmatrix} 5 \\ -1 \\ 2 \end{bmatrix} \quad (5.7)$$

抽出された係数行列を表示します.

```
> A; b;
```

$$\begin{bmatrix} 1 & 1 & -1 \\ 2 & 2 & 0 \\ 0 & 1 & 3 \end{bmatrix} \quad \begin{bmatrix} 5 \\ -1 \\ 2 \end{bmatrix} \quad (5.8)$$

連立1次方程式の係数行列を引数として解を求める LinearSolve コマンドを使用します.

```
> sol1 := LinearSolve(A, b);
```

$$sol1 := \begin{bmatrix} -19 \\ \frac{37}{2} \\ -\frac{11}{2} \end{bmatrix} \quad (5.9)$$

$A.v = b$ の式から$v = A^{-1}.b$を計算して解を求めます. MatrixInverse コマンドは逆行列を計算します.

```
> Sol2 := MatrixInverse(A).b;
```

$$Sol2 := \begin{bmatrix} -19 \\ \frac{37}{2} \\ -\frac{11}{2} \end{bmatrix} \quad (5.10)$$

ガウス・ジョルダン消去法を用いた解法

係数拡大行列 M を作成します．同じコマンドにオプションを追加します．

```
> M1 := GenerateMatrix( eqs, v, augmented=true );
```

$$M1 := \begin{bmatrix} 1 & 1 & -1 & 5 \\ 2 & 2 & 0 & -1 \\ 0 & 1 & 3 & 2 \end{bmatrix} \quad \textbf{(5.11)}$$

（参考） 行列 **(5.11)** は，以下の方法でも得ることができます．

$$A, b := \begin{bmatrix} 1 & 1 & -1 \\ 2 & 2 & 0 \\ 0 & 1 & 3 \end{bmatrix}, \begin{bmatrix} 5 \\ -1 \\ 2 \end{bmatrix}$$

```
> <A|b>;
```

$$\begin{bmatrix} 1 & 1 & -1 & 5 \\ 2 & 2 & 0 & -1 \\ 0 & 1 & 3 & 2 \end{bmatrix} \quad \textbf{(5.12)}$$

ガウス・ジョルダンの消去法を用いた解法になります．

```
> M2 := ReducedRowEchelonForm( M1 );
```

$$M2 := \begin{bmatrix} 1 & 0 & 0 & -19 \\ 0 & 1 & 0 & \frac{37}{2} \\ 0 & 0 & 1 & -\frac{11}{2} \end{bmatrix} \quad \textbf{(5.13)}$$

連立 1 次方程式を視覚化するコマンドを使用するために，学生向けの線形代数（LinearAlgebra）パッケージのコマンドを使用してみます．

```
> with(Student[LinearAlgebra]):
```

連立 1 次方程式を eqs として再定義します．

```
> eqs;
```

$$[x + y - z = 5, 2x + 2y = -1, y + 3z = 2] \quad \textbf{(5.14)}$$

LinearSystemPlot コマンドを使用します．各方程式は，それぞれ平面の式を表しています．

```
> LinearSystemPlot( eqs );
```

A System of Linear Equations



（参考）

以下の連立方程式を LinearSystemPlot コマンドで解きます.

$x + y = 1$

$2x - 3y = \frac{1}{2}$

```
> LinearSystemPlot( [ x+y=1, 2*x-3*y=1/2 ] );
```


A System of Linear Equations
1.2
1
0.8
0.6
0.4
0.2
-0.2
0
0.2
0.4
0.6
0.8
1.0
1.2
1.4
1.6
-0.2
-0.4
-0.6

ステップ･バイ･ステップ式 **はじめての Maple** 応用編

微分方程式に挑む！

微分方程式の定義方法およびその解法を習得します.

目次

数学モデルの作り方

1. 現実モデルを定式化する
2. モデルのための仮定を立てる
3. 数学問題を定式化する
4. 数学問題を解く
5. 解の意味を説明する
6. モデルの妥当性を検証する（必要に応じて２に戻る）
7. モデルを用いて説明，予測，決定，計画を行う

参考図書：『微分方程式で数学モデルを作ろう』，日本評論社，D・バージェス／M・ポリー著，垣田高夫／大町比佐栄訳

常微分方程式の定義と解析解

少しだけ常微分方程式のこと

関数の微分である導関数を含む方程式を微分方程式と呼び，特に独立変数がひとつの微分方程式を常微分方程式と呼びます．例えば， y(x) や x(t) などの関数が微分され，それらで式が組み立てられます．

$$\frac{dy}{dx}, \frac{d^2y}{dx^2}, \frac{dx}{dt}, \frac{d^2x}{dt^2}, \cdots\cdots$$

（参考） 常微分方程式の英語は Ordinary Differential Equation になります．通常，頭文字から ODE と呼ばれます．

１階常微分方程式

方程式に含まれる導関数の中でもっとも高い次数が１の場合（階数が１階である場合），１階常微分方程式と呼びます．通常，テキストなどでは，以下のように１階常微分方程式が記述されています．

$$\frac{dy}{dx} = \left(1 - y^2\right) \tan x$$

あるいは，

$$y' = \left(1 - y^2\right) \tan x$$

ただし，もう少し記述を正確にすると，以下のようになります．

$$y'(x) = \left(1 - y(x)^2\right) \tan x$$

つまり， y は x の関数となります．

ここで，初期化します．

```
> restart;
```

（例１） $\frac{d}{dx} y(x) = \left(1 - y(x)^2\right) \tan(x)$

１階の常微分方程式 $\frac{d}{dx} y(x) = \left(1 - y(x)^2\right) \tan(x)$ を定義します．

```
> ode1 := diff(y(x),x) = (1-y(x)^2)*tan(x);
```

$$ode1 := y'(x) = \left(1 - y(x)^2\right) \tan(x) \qquad \textbf{(2.2.1.1)}$$

教科書などでは，変数分離法を用いて解きますが，Maple は dsolve コマンドで解きます．

任意定数（ _C1 ）を含む一般解は，以下のようになります．

```
> sol1 := dsolve(ode1);
```

$$sol1 := y(x) = \tanh(-\ln(\cos(x)) + \_C1) \qquad \textbf{(2.2.1.2)}$$

任意定数 _C をパラメータ p に置き換えます．

```
> sol1a := subs( _C1 = p, sol1 );
```

$$sol1a := y(x) = \tanh(-\ln(\cos(x)) + p) \qquad \textbf{(2.2.1.3)}$$

式 $sol1a := y(x) = \tanh(-\ln(\cos(x)) + p)$ の右辺を eq1 に割り当てます．

```
> eq1 := rhs( sol1a );
```

$$eq1 := \tanh(-\ln(\cos(x)) + p) \qquad \textbf{(2.2.1.4)}$$

plots パッケージの animate コマンドを用いて，パラメータ p の変化を見てみます．

```
> plots[animate](plot, [ eq1, x=-5..5 ], p=-10..10 );
```



（参考）
アニメーションの操作は，以下の操作パネルから行うことができます．
操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．

（例２） $x\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right)=y(x)+\sqrt{x^2+y(x)^2}$

1 階の常微分方程式 $x\left(\frac{\mathrm{d}}{\mathrm{d}x}y(x)\right)=y(x)+\sqrt{x^2+y(x)^2}$ を定義します．

```
> ode2 := x*diff(y(x),x) = y(x) +sqrt(x^2+y(x)^2);
```

$$ode2 := x\,y'(x)=y(x)+\sqrt{x^2+y(x)^2} \qquad \textbf{(2.2.2.1)}$$

ode2 の一般解を求めます．

```
> sol2 := dsolve(ode2);
```

$$sol2 := \frac{y(x)}{x^2} + \frac{\sqrt{x^2 + y(x)^2}}{x^2} - \_C1 = 0 \qquad \textbf{(2.2.2.2)}$$

y(t) で式を整理します.

```
> sol2a := isolate(sol2, y(x));
```

$$sol2a := y(x) = \frac{-1 + \_C1^2 x^2}{2\ \_C1} \qquad \textbf{(2.2.2.3)}$$

任意定数 _C1 を q で置き換えます.

```
> sol2b := eval(sol2a, [ _C1 = q ]);
```

$$sol2b := y(x) = \frac{-1 + q^2 x^2}{2\ q} \qquad \textbf{(2.2.2.4)}$$

式 $sol2b := y(x) = \frac{-1 + q^2 x^2}{2\ q}$ の右辺を eq2 に割り当てます.

```
> eq2 := rhs( sol2b );
```

$$eq2 := \frac{-1 + q^2 x^2}{2\ q} \qquad \textbf{(2.2.2.5)}$$

plots パッケージの animate コマンドを用いて，パラメータ q の変化を見てみます.

```
> plots[animate](plot, [ eq2, x=-3..3], q=-10..10);
```



（例３）　$\frac{d}{dx} y(x) + 2\,y(x)\,\tan(x) = \sin(x)$

1 階の常微分方程式 $\frac{d}{dx} y(x) + 2\,y(x)\,\tan(x) = \sin(x)$ を定義します.

```
> ode3 := diff(y(x),x) + 2*y(x)*tan(x)=sin(x);
```

$$ode3 := y'(x) + 2\,y(x)\,\tan(x) = \sin(x) \qquad \textbf{(2.2.3.1)}$$

ode3 の一般解を求めます.

```
> sol3 := dsolve(ode3);
```

$$sol3 := y(x) = \frac{2\cos(2x) + 2 + 3\,\_C1\cos(x) + \_C1\cos(3x)}{4\cos(x)} \qquad \textbf{(2.2.3.2)}$$

一般解 sol3 を展開します.

```
> sol3a := expand( sol3 );
```

$$sol3a := y(x) = \cos(x) + \cos(x)^2\,\_C1 \qquad \textbf{(2.2.3.3)}$$

任意定数 _C1 を r で置き換えます.

```
> sol3b := subs( _C1=r, sol3a );
```

$$sol3b := y(x) = \cos(x) + \cos(x)^2 r \qquad \textbf{(2.2.3.4)}$$

式 $sol3b := y(x) = \cos(x) + \cos(x)^2 r$ の右辺を eq に割り当てます.

```
> eq3 := rhs( sol3b );
```

$$eq3 := \cos(x) + \cos(x)^2 r \qquad \textbf{(2.2.3.5)}$$

plots パッケージの animate コマンドを用いて，パラメータ r の変化を見てみます.

```
> plots[animate](plot, [ eq3, x=-2*Pi..2*Pi], r=-10..10);
```



２階常微分方程式の解

方程式に含まれる導関数の中でもっとも高い次数が２の場合（階数が２階である場合），２階常微分方程式と呼びます．

1. 解を理解するためのポイント①

解の形が

$$x(t) = C \cdot \mathrm{e}^{\lambda t}$$

と仮定されます．ただし，λ は定数になります．

2. 解を理解するためのポイント②

λ_1 と λ_2 の組は以下の３つの型に分けられます．

1）異なる２つの実数になる場合

2）異なる２つの複素数になる場合

3）同じ二つの実数になる場合

3. 解を理解するためのポイント③

重ね合わせの原理によって，解は以下のように記述できます（ただし，C_1 と C_2 は任意常数）．

$$x(t) = C_1 \cdot \mathrm{e}^{\lambda_1 t} + C_2 \cdot \mathrm{e}^{\lambda_2 t}$$

１自由度粘性減衰振動系の解

以上のポイントを踏まえて，以下の減衰する振動（１自由度粘性減衰振動系）を考えます．



初期化します．

```
> restart:
```

ニュートンの第２法則より，力は加速度に比例し，加速度は時間に関する位置の２階微分で表されることがわかります．以上の知見をもとに，１自由度の振動系を運動方程式で記述します．すなわち，ここでは，2階の常微分方程式を定義することになります．

（復習）方程式に含まれる導関数の中でもっとも高い次数が２の場合（階数が２階である場合），２階常微分方程式と呼びます．

```
> deq1 := m*diff(x(t), t, t) + c*diff(x(t), t) + k*x(t) = 0;
```

$$deq1 := m\,\ddot{x}(t) + c\,\dot{x}(t) + k\,x(t) = 0 \qquad \textbf{(2.3.1.1)}$$

解を理解するためのポイント①より

方程式の一般解を以下のような指数関数型で仮定します（ C は任意定数 ）．

```
> eq1 := x(t) = C*exp(lambda*t);
```

$$eq1 := x(t) = C\,e^{\lambda t} \qquad \textbf{(2.3.1.2)}$$

続けて，式 $deq1 := m\,\ddot{x}(t) + c\,\dot{x}(t) + k\,x(t) = 0$ に $eq1 := x(t) = C\,e^{\lambda t}$ 代入します．

```
> eq2 := eval(deq1, [eq1]);
```

$$eq2 := m\,C\,\lambda^2\,e^{\lambda t} + c\,C\,\lambda\,e^{\lambda t} + k\,C\,e^{\lambda t} = 0 \qquad \textbf{(2.3.1.3)}$$

次に，式 $eq2 := m\,C\,\lambda^2\,e^{\lambda t} + c\,C\,\lambda\,e^{\lambda t} + k\,C\,e^{\lambda t} = 0$ の両辺を $C\,e^{\lambda t}$ で割ります．

```
> eq2a := eq2/(C*exp(lambda*t));
```

$$eq2a := \frac{m\,C\,\lambda^2\,e^{\lambda t} + c\,C\,\lambda\,e^{\lambda t} + k\,C\,e^{\lambda t}}{C\,e^{\lambda t}} = 0 \qquad \textbf{(2.3.1.4)}$$

最後に，式 $eq2a := \frac{m\,C\,\lambda^2\,e^{\lambda t} + c\,C\,\lambda\,e^{\lambda t} + k\,C\,e^{\lambda t}}{C\,e^{\lambda t}} = 0$ を展開します．

```
> eq2b := expand(eq2a);
```

$$eq2b := m\,\lambda^2 + c\,\lambda + k = 0 \qquad \textbf{(2.3.1.5)}$$

この式 $eq2b := m\,\lambda^2 + c\,\lambda + k = 0$ は特性方程式と呼ばれ，その根は特性根と呼ばれます．ここで，特性方程式 $eq2b := m\,\lambda^2 + c\,\lambda + k = 0$ を λ について解きます．また，２つの特性根を λ_1, λ_2 に，そのまま割り当てます．

```
> (lambda[1], lambda[2]) := solve(eq2b, lambda);
```

$$\lambda_1, \lambda_2 := \frac{-c + \sqrt{c^2 - 4\,m\,k}}{2\,m}, -\frac{c + \sqrt{c^2 - 4\,m\,k}}{2\,m} \qquad \textbf{(2.3.1.6)}$$

特性根 λ_1, λ_2 の性質は，$c^2 - 4\,m\,k$ の符号によって変わります．すなわち；

1) $c^2 - 4\,m\,k > 0$ のとき，異なる２つの実根を持ちます．

2) $c^2 - 4\,m\,k < 0$ のとき，異なる２つの虚根を持ちます．

3) $c^2 - 4\,m\,k = 0$ のとき，重根（実数）を持ちます．

（参考）これは，高校数学で習う判別式を表しています．判別式とは，２次方程式が異なる２つの実根を持つかどうかを判別するための式になります．

以上から，**解を理解するためのポイント②**

λ_1 と λ_2 の組は以下の３つの型に分けられます．

１）異なる２つの実数になる場合

２）異なる２つの複素数になる場合

３）同じ二つの実数になる場合

は，以下のような表現に改めることができます．

特性方程式 $m\lambda^2 + c\lambda + k = 0$ は，特性根 λ_1 と λ_2 の性質によって３つの型に分けられます．

１）異なる２つの実根を持つ場合（$c^2 - 4\,m\,k > 0$）

２）異なる２つの虚根を持つ場合（$c^2 - 4\,m\,k < 0$），そして

３）重根（実数）を持つ場合（$c^2 - 4\,m\,k = 0$）

ここで，**解を理解するためのポイント③**

重ね合わせの原理から，式 $eq2b := m\lambda^2 + c\lambda + k = 0$ の一般解は，はじめに仮定した $eq1 := x(t) = C\,e^{\lambda t}$ より，以下のように求められます．

$$x(t) = C1\,e^{\lambda_1 t} + C2\,e^{\lambda_2 t}$$

ただし，C1，C2 は任意定数です．

以上を sol1 として定義します．

```
> sol1 := C1*exp( lambda[1] * t) + C2*exp( lambda[2] * t);
```

$$sol1 := C1\,e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} + C2\,e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}} \qquad \textbf{(2.3.1.7)}$$

同様に，dsolve コマンドを用いて，はじめの２階常微分方程式 $deq1 := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0$ の解を求めてみます．

```
> dsolve(deq1);
```

$$x(t) = \_C1\,e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} + \_C2\,e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}} \qquad \textbf{(2.3.1.8)}$$

ひとつずつ式を変形しながら求めた解

$$sol1 := C1\,e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} + C2\,e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}}$$

と同じになります．

この一般解は，特性方程式の３つの型（特性根が実根，虚根，あるいは重根）によって，それぞれ異なる挙動を示します．

1）異なる２つの実根を持つ場合（$c^2-4\,m\,k\ >0$のとき）

2）異なる２つの虚根を持つ場合（$c^2-4\,m\,k\ <0$のとき）

3）重根（実数の根）を持つ場（$c^2-4\,m\,k\ =0$のとき）

指数関数の特性

初期化します.

```
> restart:
```

指数関数を f として以下のように定義します.

```
> f := exp(lambda1*t) + exp(lambda2*t);
```

$$f := e^{\lambda 1\, t} + e^{\lambda 2\, t} \qquad \textbf{(2.3.2.1)}$$

1） 異なる２つの実根を持つ場合

a) ともに正の場合

特性根を定義します.

```
> pa := [lambda1 = 1, lambda2 = 2];
```

$$pa := [\lambda 1 = 1, \lambda 2 = 2] \qquad \textbf{(2.3.2.1.1.1)}$$

```
> fa := eval(f, pa);
```

$$fa := e^{t} + e^{2\, t} \qquad \textbf{(2.3.2.1.1.2)}$$

時間が十分経った（ t → +∞ ）としてプロットします.

```
> plot(fa, t=0..+infinity);
```



+∞ に発散します.

特性根を複素数として定義し，複素平面上にプロットしてみます.

```
> va := [1 + 0*I, 2 + 0*I];
```

$$va := [1, 2] \qquad \textbf{(2.3.2.1.1.3)}$$

```
> plots[complexplot](va, -2..2, style=point, symbolsize=24);
```



実軸上にプロットされます.

b) 正と負の場合

特性根を定義します.

```
> pb := [lambda1 = -1, lambda2 = 1];
```

$$pb := [\lambda 1 = -1, \lambda 2 = 1] \qquad \textbf{(2.3.2.1.2.1)}$$

```
> fb := eval(f, pb);
```

$$fb := e^{-t} + e^{t} \qquad \textbf{(2.3.2.1.2.2)}$$

時間が十分経った（ t → +∞ ）としてプロットします.

```
> plot(fb, t=0..+infinity);
```



やはり発散します.

特性根を複素数として定義し，複素平面上にプロットしてみます.

```
> vb := [-1 + 0*I, 1 + 0*I];
```

$$vb := [-1, 1] \qquad \textbf{(2.3.2.1.2.3)}$$

```
> plots[complexplot](vb, -2..2, style=point, symbolsize=24);
```



実軸上にプロットされます.

c) ともに負の場合

特性根を定義します.

```
> pc := [lambda1 = -1, lambda2 = -2];
```

$$pc := [\lambda 1 = -1, \lambda 2 = -2] \qquad \textbf{(2.3.2.1.3.1)}$$

```
> fc := eval(f, pc);
```

$$fc := e^{-t} + e^{-2t} \qquad \textbf{(2.3.2.1.3.2)}$$

時間が十分経った（ t → +∞ ）としてプロットします.

```
> plot(fc, t=0..+infinity);
```



収束します.

特性根を複素数として定義し，複素平面上にプロットしてみます.

```
> vc := [-1 + 0*I, -2 + 0*I];
```

$$vc := [-1, -2] \qquad \textbf{(2.3.2.1.3.3)}$$

```
> plots[complexplot](vc, -2..2, style=point, symbolsize=24);
```



実軸上にプロットされます.

2) 異なる２つの虚根を持つ場合

d) 実部が正の場合

特性根を定義します

```
> pd := [lambda1 = 1+I, lambda2 = 1-I];
```

$$pd := [\lambda 1 = 1 + \mathrm{I}, \lambda 2 = 1 - \mathrm{I}] \qquad \textbf{(2.3.2.2.1.1)}$$

```
> fd1 := eval(f, pd);
```

$$fd1 := e^{(1+\mathrm{I})\,t} + e^{(1-\mathrm{I})\,t} \qquad \textbf{(2.3.2.2.1.2)}$$

```
> fd2 := evalc(fd1);
```

$$fd2 := 2\,e^{t}\cos(t) \qquad \textbf{(2.3.2.2.1.3)}$$

時間が十分経った（ t → +∞ ）としてプロットします.

```
> plot(fd2, t=0..+infinity);
```



特性根を複素平面上にプロットしてみます.

```
> vd := [1 + 1*I, 1 - 1*I];
```

$$vd := [1 + \mathrm{I}, 1 - \mathrm{I}] \qquad \textbf{(2.3.2.2.1.4)}$$

```
> plots[complexplot](vd, -2..2, style=point, symbolsize=24);
```



右半平面にプロットされます.

e) 実部が負の場合

特性根を定義します

```
> pe := [lambda1 = -1+I, lambda2 = -1-I];
```

$$pe := [\lambda 1 = -1 + I, \lambda 2 = -1 - I] \quad \textbf{(2.3.2.2.2.1)}$$

```
> fe1 := eval(f, pe);
```

$$fe1 := e^{(-1+I)t} + e^{(-1-I)t} \quad \textbf{(2.3.2.2.2.2)}$$

```
> fe2 := evalc(fe1);
```

$$fe2 := 2\,e^{-t}\cos(t) \quad \textbf{(2.3.2.2.2.3)}$$

時間が十分経った（ t → +∞ ）としてプロットします.

```
> plot(fe2, t=0..+infinity);
```



特性根を複素平面上にプロットしてみます.

```
> ve := [-1 + 1*I, -1 - 1*I];
```

$$ve := [-1 + I, -1 - I] \qquad \textbf{(2.3.2.2.2.4)}$$

```
> plots[complexplot](ve, -2..2, style=point, symbolsize=24);
```



左半平面にプロットされます.

少しだけ考察

特性根の実部の符号が正のとき発散し，負のとき収束します．また，虚部があるばあい振動的になります.

２階常微分方程式の行列表現

初期化します．

```
> restart:
```

２階常微分方程式を行列表現に変換します．

```
> deq1 := m*diff(x(t), t, t) + c*diff(x(t), t) + k*x(t) = 0;
```

$$deq1 := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0 \qquad \textbf{(2.4.1)}$$

新たな時間で変化する（状態）変数を導入します．x(t) を時間 t について１階微分して得られる変数を v(t) とします．

```
> deq2 := diff(x(t), t) = v(t);
```

$$deq2 := \dot{x}(t) = v(t) \qquad \textbf{(2.4.2)}$$

deq1 に deq2 を代入して，deq3 とします．

```
> deq3 := eval(deq1, [deq2]);
```

$$deq3 := m\dot{v}(t) + c\,v(t) + k\,x(t) = 0 \qquad \textbf{(2.4.3)}$$

deq3 を $\frac{\mathrm{d}}{\mathrm{d}t}\,v(t)$ で整理します．

```
> deq4 := isolate(deq3, diff( v(t),t ) );
```

$$deq4 := \dot{v}(t) = \frac{-c\,v(t) - k\,x(t)}{m} \qquad \textbf{(2.4.4)}$$

deq4 を展開します．

```
> deq5 := expand(deq4);
```

$$deq5 := \dot{v}(t) = -\frac{c\,v(t)}{m} - \frac{k\,x(t)}{m} \qquad \textbf{(2.4.5)}$$

ここで，行列で式を整理するために，x(t) と v(t) でベクトル xv を定義します．

```
> xv := Vector([x(t), v(t)]);
```

$$xv := \begin{bmatrix} x(t) \\ v(t) \end{bmatrix} \qquad \textbf{(2.4.6)}$$

deq2 と deq5 を確認します．

```
> deq2;
```

$$\dot{x}(t) = v(t) \qquad \textbf{(2.4.7)}$$

```
> deq5;
```

$$\dot{v}(t) = -\frac{c\,v(t)}{m} - \frac{k\,x(t)}{m} \tag{2.4.8}$$

deq2，deq5 ともに 1 次の導関数で整理され，1 階常微分方程式が導かれます．

ここで，行列表現を用いて，以下のように deq2 および deq5 を整理します．

$$\frac{d}{dt}\begin{bmatrix} x(t) \\ v(t) \end{bmatrix} = \begin{bmatrix} 0 & 1 \\ -\frac{k}{m} & -\frac{c}{m} \end{bmatrix}\begin{bmatrix} x(t) \\ v(t) \end{bmatrix}$$

$A = \begin{bmatrix} 0 & 1 \\ -\frac{k}{m} & -\frac{c}{m} \end{bmatrix}$ として，上式を deq6 として定義します．

```
> A := Matrix( 2,2, [ 0, 1, -k/m, -c/m ]);
```

$$A := \begin{bmatrix} 0 & 1 \\ -\frac{k}{m} & -\frac{c}{m} \end{bmatrix} \tag{2.4.9}$$

```
> deq6 := map(diff, xv, t) = A.xv;
```

$$deq6 := \begin{bmatrix} \dot{x}(t) \\ \dot{v}(t) \end{bmatrix} = \begin{bmatrix} v(t) \\ -\frac{c\,v(t)}{m} - \frac{k\,x(t)}{m} \end{bmatrix} \tag{2.4.10}$$

ここで，A の固有値を求めてみます．

```
> LinearAlgebra[Eigenvalues](A);
```

$$\begin{bmatrix} \frac{-c+\sqrt{c^2-4\,m\,k}}{2\,m} \\ -\frac{c+\sqrt{c^2-4\,m\,k}}{2\,m} \end{bmatrix} \tag{2.4.11}$$

これは，特性方程式の特性根と同じ値になります．

特性方程式 $eq2b := m\,\lambda^2 + c\,\lambda + k = 0$

特性根 $\lambda_1, \lambda_2 := \frac{-c+\sqrt{c^2-4\,m\,k}}{2\,m}, -\frac{c+\sqrt{c^2-4\,m\,k}}{2\,m}$

すなわち，の一般解 $x(t) = \_C1\, e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} + \_C2\, e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}}$ の挙動は係数行列の固有値を調べることよっても判別することができます．

$deq6 := \begin{bmatrix} \dot{x}(t) \\ \dot{v}(t) \end{bmatrix} = \begin{bmatrix} v(t) \\ -\frac{c\,v(t)}{m} - \frac{k\,x(t)}{m} \end{bmatrix}$ を（制御工学の分野で）**状態方程式**と呼びます．また，A を係数行列と呼び，制御システムの理論的な設計に用いられます．

連立微分方程式の解法

deq2 := $\dot{x}(t) = v(t)$ と

deq5 := $\dot{v}(t) = -\frac{c\,v(t)}{m} - \frac{k\,x(t)}{m}$ を連立させて解いてみます．

dsolve コマンドを用いて連立微分方程式を解きます．

```
> dsolve([deq2, deq5], [x(t), v(t)]);
```

$$\left\{ v(t) = \_C1\, e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} + \_C2\, e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}}, x(t) = -\frac{1}{2k}\left(\_C1\, e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} c + \_C1\, e^{\frac{\left(-c+\sqrt{c^2-4mk}\right)t}{2m}} \sqrt{c^2-4mk} + \_C2\, e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}} c - \_C2\, e^{-\frac{\left(c+\sqrt{c^2-4mk}\right)t}{2m}} \sqrt{c^2-4mk} \right) \right\} \quad \textbf{(2.4.1.1)}$$

初期値問題

初期化します.

```
> restart;
```

２階常微分方程式を定義します.

```
> ode := m*diff(x(t), t, t) + c*diff(x(t), t) + k*x(t) = 0;
```

$$ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0 \qquad \textbf{(2.5.1)}$$

式 $ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0$ の解を求めます.

```
> dsolve(ode);
```

$$x(t) = \_C1\, e^{\frac{\left(-c+\sqrt{c^2-4km}\right)t}{2m}} + \_C2\, e^{-\frac{\left(c+\sqrt{c^2-4km}\right)t}{2m}} \qquad \textbf{(2.5.2)}$$

初期条件： $x(0) = -1,\ \dfrac{dx(0)}{dt} = 0.$

初期条件の定義には，微分演算子 D を用い，シーケンスとして定義します.

```
> ics := x(0)=-1, D(x)(0)=0;
```

$$ics := x(0) = -1,\ D(x)(0) = 0 \qquad \textbf{(2.5.3)}$$

{　} （集合の記号）を用いて，方程式と初期条件を整理します（括ります）.

適当な係数 m, c, k を定義し，方程式に代入します.

```
> params := [ m=1, c=1/3, k=11];
```

$$params := \left[m = 1,\ c = \frac{1}{3},\ k = 11\right] \qquad \textbf{(2.5.4)}$$

式 $ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0$ に $params := \left[m = 1,\ c = \frac{1}{3},\ k = 11\right]$ を代入します.

```
> odea := eval(ode, params);
```

$$odea := \ddot{x}(t) + \frac{\dot{x}(t)}{3} + 11\,x(t) = 0 \qquad \textbf{(2.5.5)}$$

初期値問題として解きます.

```
> sol := dsolve( { odea, ics } );
```

(2.5.6)

$$sol := x(t) = -\frac{\sqrt{395}\, e^{-\frac{t}{6}} \sin\left(\frac{\sqrt{395}\, t}{6}\right)}{395} - e^{-\frac{t}{6}} \cos\left(\frac{\sqrt{395}\, t}{6}\right) \quad \textbf{(2.5.6)}$$

求められた解は，特殊解と呼ばれます．

なお，引数は以下のような集合になっています．

（復習） 集合の定義：　順番は保持されません．重複する要素は省かれます．

```
> { ode, ics };
```

$$\{m\ddot{x}(t) + c\dot{x}(t) + k x(t) = 0, x(0) = -1, D(x)(0) = 0\} \quad \textbf{(2.5.7)}$$

式 $sol := x(t) = -\frac{\sqrt{395}\, e^{-\frac{t}{6}} \sin\left(\frac{\sqrt{395}\, t}{6}\right)}{395} - e^{-\frac{t}{6}} \cos\left(\frac{\sqrt{395}\, t}{6}\right)$ の右辺を eq に割り当てます．

```
> eq1 := rhs(sol);
```

$$eq1 := -\frac{\sqrt{395}\, e^{-\frac{t}{6}} \sin\left(\frac{\sqrt{395}\, t}{6}\right)}{395} - e^{-\frac{t}{6}} \cos\left(\frac{\sqrt{395}\, t}{6}\right) \quad \textbf{(2.5.8)}$$

式 $eq1 := -\frac{\sqrt{395}\, e^{-\frac{t}{6}} \sin\left(\frac{\sqrt{395}\, t}{6}\right)}{395} - e^{-\frac{t}{6}} \cos\left(\frac{\sqrt{395}\, t}{6}\right)$ をプロットします．

```
> plot( eq1, t=0..10 );
```



t = 0 のとき値は，x(0) = -1 になっています．

（参考） 自由振動の原理



初期条件をパラメータとして定義しなおします．

```
> pics := x(0) = x0, D(x)(0) = 0;
```

$$pics := x(0) = x0,\ \mathrm{D}(x)(0) = 0 \qquad \textbf{(2.5.9)}$$

$pics := x(0) = x0,\ \mathrm{D}(x)(0) = 0$ の初期条件で方程式 $ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0$ を解きます.

```
> psol := dsolve( { odea, pics } );
```

$$psol := x(t) = \frac{x0\sqrt{395}\ \mathrm{e}^{-\frac{t}{6}}\sin\left(\frac{\sqrt{395}\ t}{6}\right)}{395} + x0\,\mathrm{e}^{-\frac{t}{6}}\cos\left(\frac{\sqrt{395}\ t}{6}\right) \qquad \textbf{(2.5.10)}$$

式 $psol := x(t) = \frac{x0\sqrt{395}\ \mathrm{e}^{-\frac{t}{6}}\sin\left(\frac{\sqrt{395}\ t}{6}\right)}{395} + x0\,\mathrm{e}^{-\frac{t}{6}}\cos\left(\frac{\sqrt{395}\ t}{6}\right)$ の右辺を eq2 に割り当てます.

```
> eq2 := rhs(psol);
```

$$eq2 := \frac{x0\sqrt{395}\ \mathrm{e}^{-\frac{t}{6}}\sin\left(\frac{\sqrt{395}\ t}{6}\right)}{395} + x0\,\mathrm{e}^{-\frac{t}{6}}\cos\left(\frac{\sqrt{395}\ t}{6}\right) \qquad \textbf{(2.5.11)}$$

初期値を変化させて，アニメーションを作成します.

```
> plots[animate](plot, [ eq2, t=0..10], x0=-1..1 );
```

$$x0 = -1.$$



（参考）

アニメーションの操作は，以下の操作パネルから行うことができます．

操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．

ラプラス変換による初期値問題の解法

初期化します．

```
> restart:
```

積分変換は，数学的に（あるいは物理的に）関数 $f(t)$ をある領域 t から別の領域 α へ変換する積分です．

$$g(\alpha) = \int_a^b f(t)\, K(\alpha, t)\, dt$$

積分変換として，ラプラス変換やフーリエ変換がよく知られています．例えば，ラプラス変換は，微分方程式を代数方程式に変換します（計算が簡単になります）．また，フーリエ変換は，時系列データを時間領域から周波数領域へ変換します（見えないものが見えてきます）．

積分変換を利用するには，inttrans パッケージのコマンドを使用します．

```
> with(inttrans);
```

[*addtable, fourier, fouriercos, fouriersin, hankel, hilbert, invfourier, invhilbert, invlaplace, invmellin, laplace, mellin, savetable*] **(2.6.1)**

ラプラス変換を用いて微分方程式の初期値問題を解いてみます．

ここで，２階の常微分方程式を定義します．

```
> deq := diff(x(t),t,t) + 2*diff(x(t),t) - 3*x(t) = 0;
```

$$deq := \ddot{x}(t) + 2\dot{x}(t) - 3x(t) = 0 \quad \textbf{(2.6.2)}$$

初期値 $x(0) = 0$, $\frac{d}{dt}x(0) = 1$ を集合 { } として定義します．

（参考）D は微分演算子になります．

```
> ics := { x(0)=0, D(x)(0)=1 };
```

$$ics := \{x(0) = 0, D(x)(0) = 1\} \quad \textbf{(2.6.3)}$$

微分方程式をラプラス変換します（領域 t から領域 s に変換します）．

```
> Leq1 := laplace(deq, t, s);
```

$$Leq1 := s^2\, laplace(x(t), t, s) - D(x)(0) - s\,x(0) + 2\,s\, laplace(x(t), t, s) - 2\,x(0) - 3\, laplace(x(t), t, s) = 0 \quad \textbf{(2.6.4)}$$

式を見やすくするために，いったん laplace(x(t),t,s) を X に置き換えます．

```
> Leq2 := subs( laplace(x(t),t,s)=X, Leq1 );
```

$$Leq2 := s^2 X - \mathrm{D}(x)(0) - s\,x(0) + 2\,s\,X - 2\,x(0) - 3\,X = 0 \qquad \textbf{(2.6.5)}$$

初期条件を式に代入・評価します.

```
> Leq3 := eval(Leq2, ics);
```

$$Leq3 := s^2 X - 1 + 2\,s\,X - 3\,X = 0 \qquad \textbf{(2.6.6)}$$

X で式を整理します.

```
> Leq4 := isolate(Leq3, X);
```

$$Leq4 := X = \frac{1}{s^2 + 2\,s - 3} \qquad \textbf{(2.6.7)}$$

X を元の laplace(x(t),t,s) に置き換えます.

```
> Leq5 := subs( X=laplace(x(t),t,s), Leq4 );
```

$$Leq5 := laplace(x(t), t, s) = \frac{1}{s^2 + 2\,s - 3} \qquad \textbf{(2.6.8)}$$

右辺を部分分数に変換します.

```
> Leq6 := convert(Leq5, parfrac);
```

$$Leq6 := laplace(x(t), t, s) = -\frac{1}{4\,(s+3)} + \frac{1}{4\,(s-1)} \qquad \textbf{(2.6.9)}$$

式 $Leq6 := laplace(x(t), t, s) = -\frac{1}{4\,(s+3)} + \frac{1}{4\,(s-1)}$ を**逆**ラプラス変換します(s --> t)
.

```
> Leq7 := invlaplace(Leq6, s, t);
```

$$Leq7 := x(t) = -\frac{\mathrm{e}^{-3t}}{4} + \frac{\mathrm{e}^{t}}{4} \qquad \textbf{(2.6.10)}$$

初期値問題の解になります.

（参考）以下は，dsolve コマンドを用いた解になります.

```
> dsol := dsolve({deq} union ics);
```

$$dsol := x(t) = -\frac{\mathrm{e}^{-3t}}{4} + \frac{\mathrm{e}^{t}}{4} \qquad \textbf{(2.6.11)}$$

（参考）union は集合の和（演算子）になります.

```
> {deq} union ics;
```

$$\{\ddot{x}(t) + 2\,\dot{x}(t) - 3\,x(t) = 0,\ x(0) = 0,\ \mathrm{D}(x)(0) = 1\} \qquad \textbf{(2.6.12)}$$

2 階常微分方程式の数値解法

初期化します.

```
> restart;
```

２階の常微分方程式を定義します.

```
> ode := m*diff(x(t), t, t) + c*diff(x(t), t) + k*x(t) = 0;
```

$$ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0 \tag{3.1}$$

パラメータを定義します.

```
> params := [ m=1, c=1/3, k=11];
```

$$params := \left[m = 1, c = \frac{1}{3}, k = 11\right] \tag{3.2}$$

パラメータ $params := \left[m = 1, c = \frac{1}{3}, k = 11\right]$ を式 $ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0$ に代入します.

```
> odea := eval(ode, params);
```

$$odea := \ddot{x}(t) + \frac{\dot{x}(t)}{3} + 11\,x(t) = 0 \tag{3.3}$$

初期条件を定義します.

```
> ics := x(0)=-1, D(x)(0)=0;
```

$$ics := x(0) = -1, \mathrm{D}(x)(0) = 0 \tag{3.4}$$

数値解を求めるためのオプション（**numeric**）を指定します.

```
> dsol := dsolve( {odea,ics}, numeric );
```

$$dsol := \mathbf{proc}(x\_rkf45)\ \ldots\ \mathbf{end\ proc} \tag{3.5}$$

数値解を求める処理が一種のプログラム形式で出力されます．すなわち，dsol にある値を引数として与えることで，その値に対する出力値が計算されます．このプログラム形式の数値解を，**解関数**と呼びます．さらに，Maple の中では一般的な名称として，**プロシージャ**（処理の単位）という言葉が使用されます.

t=0 として，解関数 dsol を計算してみます.

```
> dsol(0);
```

$$[t = 0., x(t) = -1., \dot{x}(t) = 0.] \tag{3.6}$$

t=0 のときの $x(0)$ と $\frac{d}{dt}x(0)$の値が計算されます.

同様に t=1 のときの値を計算してみます.

```
> dsol(1);
```

$$[t = 1., x(t) = 0.841399896866334429, \dot{x}(t) = -0.477906337087002476] \quad \textbf{(3.7)}$$

以下は，t=Pi のときの値です.

```
> dsol(Pi);
```

$$[t = 3.14159265358979, x(t) = 0.353992482077243486, \dot{x}(t) = -1.63544709648800524] \quad \textbf{(3.8)}$$

微分方程式の解関数は，plots パッケージの odeplot コマンドを用いて，横軸と縦軸を指定しプロットすることができます.

```
> plots[odeplot](dsol, [t, x(t)], 0..10);
```



numpoints=1000 を指定することで，多少なめらかな線になります.

```
> plots[odeplot](dsol, [t, x(t)], 0..10, numpoints=1000);
```



オプション frames でフレーム数を指定することによって，アニメーションを作成することができます.

```
> plots[odeplot](dsol, [t, x(t)], 0..10, numpoints=1000, frames=52);
```



（参考）

アニメーションの操作は，以下の操作パネルから行うことができます．
操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．

縦軸，横軸にそれぞれ位置 $x(t)$，速度 $\frac{d}{dt}x(t)$ を取ったときのアニメーションになります．

```
> plots[odeplot](dsol, [x(t), diff(x(t),t)], 0..10, numpoints=1000,
  frames=24);
```


3
2
x'
1
-1
-0.8
-0.6
-0.4
-0.2
0
0.2
0.4
0.6
0.8
x
-1
-2

ステップ･バイ･ステップ式 **はじめての Maple** 応用編

プログラミング前夜

Maple プログラミング言語の基礎を習得します.

目次

処理の再利用

プロシージャ化（処理を１行で記述可能な場合）

（基本構文）

```
プロシージャ名 := ( 変数 ) -> コマンド（変数を含む処理を記述）;
```

- 変数はプロシージャの引数になります.
- ->（アロー演算子）を使用します.

初期化します.

```
> restart:
```

引数がひとつの場合

```
> myproc1 := x -> x^2;
```

$$myproc1 := x \mapsto x^2 \qquad \textbf{(1.1.1)}$$

プロシージャに値を与えて実行します．入力方法，実行方法は通常のコマンドと同じです.

```
> myproc1(2);
```

$$4 \qquad \textbf{(1.1.2)}$$

変数（引数）がふたつの場合になります.

```
> myproc2 := (x, y) -> x^2 + y^2;
```

$$myproc2 := (x, y) \mapsto x^2 + y^2 \qquad \textbf{(1.1.3)}$$

```
> myproc2(3, 4);
```

$$25 \qquad \textbf{(1.1.4)}$$

変数（引数）なしのプロシージャになります.

```
> myproc3 := ( ) -> plot(sin(x), x=-Pi..Pi);
```

$$myproc3 := ( ) \mapsto plot(\sin(x), x = -\pi..\pi) \qquad \textbf{(1.1.5)}$$

```
> myproc3();
```



（参考）
プロットの変更や操作は，以下の操作パネルから行うことができます．
操作パネルは，プロットされたグラフをクリックするとツールバーの下に表示されます．



以降，１行で記述可能なプロシージャを便宜上「**ファンクション**」を呼ぶことにします．これは，次のプロシージャと区別するためになります．

プロシージャ化（処理を複数行で記述する場合）

（基本構文）

```
プロシージャ名 := proc( 引数 )
   処理;
end proc;
```

一般的に，処理は複数行で記述します．

初期化します．

```
> restart;
```

引数なしの場合

```
> myproc1 := proc()
      print( "Hello World!" );
   end proc;
```

$$myproc1 := \mathbf{proc}(\,)\ print(\text{"Hello World!"})\ \mathbf{end\ proc} \qquad \textbf{(1.2.1)}$$

プロシージャを実行してみます．

```
> myproc1;
```

$$myproc1 \qquad \textbf{(1.2.2)}$$

ただし，プロシージャ名が表示されるだけです．プロシージャを実行するためには( ) が必要になります．

```
> myproc1();
```

$$\text{"Hello World!"} \qquad \textbf{(1.2.3)}$$

プロシージャを１行で定義することも可能です．

```
> myproc2 := proc() printf( "Hello World!" ); end proc;
```

$$myproc2 := \mathbf{proc}(\,)\ printf(\text{"Hello World!"})\ \mathbf{end\ proc} \qquad \textbf{(1.2.4)}$$

プロシージャを実行します．

```
> myproc2();
Hello World!
```

ある範囲（区間）で $\sin(x)$ をプロットするプロシージャ

```
> myproc3 := proc(a, b)

    plot( sin(x), x=a..b );

end proc;
```

$$myproc3 := \mathbf{proc}(a, b)\ \ plot(\sin(x), x = a..b)\ \ \mathbf{end\ proc} \qquad \textbf{(1.2.5)}$$

-Pi..Pi の範囲でプロットします.

```
> myproc3(-Pi, Pi);
```



ある関数（ sin(x) ）を積分し，その結果（ -cos(x) ）とともに，もとの（被積分）関数（ sin(x) ）を同時にプロットするプロシージャ

```
> myproc4 := proc(f, a, b)

    # f と int(f, x) をプロット
    plot( [f, int(f, x)], x=a..b );

end proc:
```

`end proc:` のように，最後をコロン（ `:` ）で閉じると，他の処理と同じように結果が表示されません.

```
> myproc4(sin(x), -Pi, Pi);
```



グローバル変数とローカル変数

一般的に，グローバル変数は，すべてのスコープ（変数の参照範囲）から参照可能な変数（大域的に参照可能な変数）になり，ローカル変数は，そのプロシージャ内でのみ参照可能な変数（局所的に参照可能な変数）になります．

処理の内容によって，グローバル変数とローカル変数を使い分けることになりますが，通常［プロシージャの作成において］グローバル変数の使用はできるだけ避けるようにします．これは，グローバル変数に格納されている値や式が，全体の処理の中で意図せず変更される可能性を防ぐためです．



初期化します．

```
> restart;
```

（例）　グローバル変数とローカル変数の違い

プロシージャ名を myproc5 として，以下のプロシージャを定義します．

はじめに，変数 X に 5 を割り当てます（格納します）．この X は，グローバル変数になります．つまり，スコープ（参照の有効範囲）はワークシート内すべてになります．

```
> X := 5;
```

$$X := 5 \qquad \textbf{(1.3.1.1)}$$

プロシージャの先頭に，コメントとして処理の説明を記述します．シャープ（ # ）の後ろは，Maple の処理には含まれません（コメントとして無視されます）．

（参考）ローカル変数の定義
ローカル変数は，プロシージャ内の先頭において，local を用いて定義します（あるいは，宣言します）．複数定義する場合は，コンマ（ , ）で変数を区切ります．

（例）`local a, b, c, d, e, f, abc, xyz;`

```
> # ---------------------------------
  # X, Y, Z の内容を表示するプロシージャ
  # ---------------------------------
  myproc5 := proc()

     # ローカル変数の宣言
     local Y, Z;

     # 値の代入
     Y := 6;
     Z := 15;

     # 変数の内容を表示
     print( X, Y, Z );

  end proc: # myproc5
```

Y および Z は，ローカル変数になります．スコープ（参照の有効範囲）は，myproc5 プロシージャ内だけで参照可能となります．例えば，Y を表示してます．

```
> Y;
```

$$Y \qquad \textbf{(1.3.1.2)}$$

変数 Y には，何も格納されていません．

ここで，Y と Z にそれぞれ 33 と 47 を割り当てて，myproc5 を実行してみます．

```
> Y := 33;
  Z := 47;
```

$$Y := 33$$
$$Z := 47 \qquad \textbf{(1.3.1.3)}$$

```
> myproc5();
```

$$5, 6, 15 \qquad \textbf{(1.3.1.4)}$$

プロシージャ内の Y と Z には，何も影響しません．

引数の型・変数の型

プロシージャ，引数，およびローカル変数の型を明示的に与えることができます．

（基本構文）

```
プロシージャ名 := proc( 引数::型 )::型
   local 変数::型
  処理;
end proc;
```

2 つの整数値（integer）を足し合わせ，計算結果も整数値（integer）とします．

```
> # ------------------------------
  # 整数値の足し算
  # ------------------------------
  myAdd := proc( a::integer, b::integer )::integer;

     # ローカル変数の宣言
     local sum;

     # 引数 a と b の足し算
     sum := a + b;

  end proc: # myAdd
```

1 と 2 の足し算を実行し，変数 p に割り当てます（格納します）．

```
> p := myAdd(1, 2);
```

$$p := 3 \tag{1.4.1}$$

変数 p の型を確認します．

```
> whattype(p);
```

$$integer \tag{1.4.2}$$

数値 1.0 （浮動小数点型＝*float*） と整数値 2 （整数型＝*integer*） の足し算を実行します．

```
> p := myAdd(1.0, 2);
Error, invalid input: myAdd expects its 1st argument, a, to be of
type integer, but received 1.0
```

エラーメッセージが出力されます．第 1 番目の引数（argument） = 1.0 が数値（浮動小数点型）のため，整数型と一致しません．

制御構文

処理の流れを制御する構文です.

条件分岐（if-then-else-end if）

条件（式）によって処理が分岐します.

（基本構文）

```
if 条件式１ then
   処理１;
elif 条件式２ then
   処理２;
else
   処理３;
end if;
```

初期化します.

```
> restart;
```

数字の大小を判別する条件分岐

数字の大小を判別する条件分岐を作成します.

はじめに，a および b に適当な（整数）値を代入します.

```
> a := 1;
```

$$a := 1 \qquad \textbf{(2.1.1.1)}$$

```
> b := 2;
```

$$b := 2 \qquad \textbf{(2.1.1.2)}$$

条件式 a > b の評価結果に応じて出力されるメッセージが変わります.

```
> if a > b then
     print( "a のほうが大きい" );
  else
```

```
   print( "b のほうが大きい" );
end if;
```

"b のほうが大きい" **(2.1.1.3)**

条件式 a > b を evalb で評価してみます.

```
> evalb( a > b );
```

false **(2.1.1.4)**

条件式を a < b として，evalb で評価してみます.

```
> evalb( a < b );
```

true **(2.1.1.5)**

次に，a と b が等しい場合の処理を追加します．この時点で，定義されている値は a=1 および b=2 になります．つまり，a < b （ 1 < 2 ）の関係にあります.

```
> if a > b then
   print( "a のほうが大きい" );
elif a = b then
   print( "a と b は等しい" );
else
   print( "b のほうが大きい" );
end if;
```

"b のほうが大きい" **(2.1.1.6)**

ここで，a と b を等しい値で定義しなおします.

```
> a := 3;
  b := 3;
```

$$a := 3$$
$$b := 3$$ **(2.1.1.7)**

```
> if a > b then
   print( "a のほうが大きい" );
elif a = b then
   print( "a と b は等しい" );
else
   print( "b のほうが大きい" );
end if;
```

"a と b は等しい" **(2.1.1.8)**

条件分岐を含むプロシージャの作成手順

上記の処理を引数 a，b としてプロシージャにまとめます．すなわち，再利用可能な処理にまとめます．ここでは，以下の手順でまとめることにします．

① プロシージャ名を myproc6 として，proc 構造を用意します．また，引数を a，b とします，

```
> myproc6 := proc( a, b )


  end proc:
```

② 上の大小を評価してメッセージを出力する if 文を挿入します．

```
> myproc6 := proc( a, b )
  if a > b then
     print( "a のほうが大きい" );
  elif a = b then
     print( "a と b は等しい" );
  else
     print( "b のほうが大きい" );
  end if:
  end proc:
```

③ if から end if; までの行を右方向に下げます（インデントします）．

（参考） 通常，その処理が，上位の処理にネストされているとき（入れ子になっているとき），インデントを設けます． 2 文字文から 4 文字程度が一般的です．ここでは，インデントを 3 文字文にしています．

```
> myproc6 := proc(a, b)
  ###
     if a > b then
     ###
        print( "a のほうが大きい" );
  ###
     elif a = b then
     ###
        print( "a と b は等しい" );
  ###
     else
     ###
        print( "b のほうが大きい" );
  ###
     end if;

  end proc:
```

④ 説明やコメントを追加します（### は取り除きます．### はインデントを示すために挿入しました．通常は必要ありません）．

```
> # ---------------------------------
  # 2変数に格納されている値の大小を比較
  # ---------------------------------
  myproc6 := proc(a, b)

     # a > b の場合
     if a > b then

        # 以下のメッセージを表示
        print( "a のほうが大きい" );

     # a = b の場合
     elif a = b then

        # 以下のメッセージを表示
        print( "a と b は等しい" );

     # 2つの条件式から外れた場合の処理
     else

        # 以下のメッセージを表示
        print( "b のほうが大きい" );

     end if;

  end proc: # myproc6
```

a と b を a > b として定義しなおします．

```
> a := 5;
  b := 3;
```

$$a := 5$$
$$b := 3 \qquad \textbf{(2.1.2.1)}$$

myproc6 に値を与えます．

```
> myproc6(a, b);
```

"a のほうが大きい" **(2.1.2.2)**

引数の変数名は a，b 以外でも使用できます．

```
> X := 30;
  Y := -20;
```

$$X := 30$$
$$Y := -20 \qquad \textbf{(2.1.2.3)}$$

同様に myproc6 に値を与えます．

```
> myproc6(X, Y);
```

"a のほうが大きい" **(2.1.2.4)**

（参考）引数を比較するプロシージャに変更

１番目の引数と２番目の引数を比較するプロシージャであると解釈しなおします.
表示内容を書き換えます.

```
> # ---------------------------------
  # ２変数に格納されている値の大小を比較
  # ---------------------------------
  myproc6 := proc(a, b)

     # a > b の場合
     if a > b then

        # 以下のメッセージを表示
        print( "引数１のほうが大きい" );

     # a = b の場合
     elif a = b then

        # 以下のメッセージを表示
        print( "引数１と引数２は等しい" );

     # ２つの条件式から外れた場合の処理
     else

        # 以下のメッセージを表示
        print( "引数２のほうが大きい" );

     end if;

  end proc: # myproc6
```

（少しだけ）一般化されたプロシージャになります.

```
> myproc6(X, Y);
```

"引数１のほうが大きい"　　**(2.1.3.1)**

繰り返し（for-do-end do）

（基本構文）

```
for インデックス名 from 開始数 by インクリメント数 to 終了数 do
   処理（インデックス名を含む）;
end do;
```

初期化します.

```
> restart;
```

足し算の実行

0, 1, 2, 3, 4 の足し算を繰り返し文を用いて実行します.

繰り返し文の前に

はじめに，ここではひとつの変数 total を使用して足し算を行い，はじめの値を 0 とします. この 0 を初期値と呼びます.

```
> total := 0;
```

$$total := 0 \qquad \textbf{(2.2.1.1.1)}$$

```
> total := total + 1;
# total := ( 0 ) + 1;
```

$$total := 1 \qquad \textbf{(2.2.1.1.2)}$$

```
> total := total + 2;
# total := ( ( 0 ) + 1 ) + 2;
```

$$total := 3 \qquad \textbf{(2.2.1.1.3)}$$

```
> total := total + 3;
# total := ( ( ( 0 ) + 1 ) + 2 ) + 3;
```

$$total := 6 \qquad \textbf{(2.2.1.1.4)}$$

```
> total := total + 4;
# total := ( ( ( ( 0 ) + 1 ) + 2 ) + 3 ) + 4;
```

$$total := 10 \qquad \textbf{(2.2.1.1.5)}$$

実際の足し算は以下のような記述で簡単に実行されます.

```
> 0 + 1 + 2 + 3 + 4;
```

$$10 \qquad \textbf{(2.2.1.1.6)}$$

ここで，インデックス `i` を導入します. とくに，i である必要はありませんが，一般的に 1 次元のときは i を使用します.

（参考）また，2 次元のときは i と j を，さらに，3 次元のときは，i，j，および k を使用します.

```
>   total := 0; # はじめの値を０とします

    i := 1;
    total :=       total   +   i;
  # total :=       ( 0 )   +   1;

    i := 2;
    total :=       total         +   i;
  # total :=     ( ( 0 ) + 1 )   +   2;

    i := 3;
    total :=       total               +   i;
  # total :=   ( ( ( 0 ) + 1 ) + 2 )   +   3;

    i := 4;
    total :=       total                     +   i;
  # total := ( ( ( ( 0 ) + 1 ) + 2 ) + 3 )   +   4;
```

$$total := 0$$
$$i := 1$$
$$total := 1$$
$$i := 2$$
$$total := 3$$
$$i := 3$$
$$total := 6$$
$$i := 4$$
$$total := 10$$ **(2.2.1.1.7)**

ここで，いったい何がなされたのでしょうか？？？

実は，

```
total := total + 1
total := total + 2
total := total + 3
total := total + 4
```

をインデックス `i` を用いて

```
total := total + i
```

と式を**一般化**しました.

繰り返し文の利用

`i` が 1 から 4 まで 1 ずつおおきくなります（これを，**インクリメント**，と表現します）．Maple の処理で記述すると以下のようになります.

```
i from 1 to 4 by 1
```

（もしくは　`i from 1 by 1 to 4` ）

さらに，Maple の繰り返し文（ `for-do-end do` 文 ）を追加します.

```
for i from 1 to 4 by 1 do

end do;
```

ここで，i の変化を見ます（ i に格納されている値を表示するだけの処理です ）．

（参考） 'i' は文字そのものを表します．

print コマンドは，表示処理を実行します．

```
> for i from 1 to 4 by 1 do

      print( 'i' = i );

  end do;
```

$$i = 1$$
$$i = 2$$
$$i = 3$$
$$i = 4$$ **(2.2.1.2.1)**

以下の処理を繰り返し文（ for-do-end do 文 ）を用いて整理します．

```
total := 0;
```

（ ここから for 文が始まり・・・ ）

```
i := 1;
total := total + i;

i := 2;
total := total + i;

i := 3;
total := total + i;

i := 4;
total := total + i;
```

（ ・・・ここで for 文が終わります）

すなわち，

for - do と end do; の間に

total := total + i を挿入し，処理を整理します．なお，total のはじめの値（初期値）は 0 になります．

```
> total := 0; # 初期値の設定

  for i from 1 to 4 by 1 do

     print( 'i' = i ); # i を参照するために残してあります
     total := total + i;
```

```
end do;
```

$$\begin{aligned} total &:= 0 \\ i &= 1 \\ total &:= 1 \\ i &= 2 \\ total &:= 3 \\ i &= 3 \\ total &:= 6 \\ i &= 4 \\ total &:= 10 \end{aligned} \qquad \textbf{(2.2.1.2.2)}$$

`i` が 1 ずつインクリメントされる場合，`by 1` を省略できます．

```
> total := 0; # 初期値の設定

  for i from 1 to 4 do

     print( 'i' = i ); # i を参照するために残してあります
     total := total + i;

  end do;
```

$$\begin{aligned} total &:= 0 \\ i &= 1 \\ total &:= 1 \\ i &= 2 \\ total &:= 3 \\ i &= 3 \\ total &:= 6 \\ i &= 4 \\ total &:= 10 \end{aligned} \qquad \textbf{(2.2.1.2.3)}$$

プロシージャへの拡張

ここで，0, 1, 2, 3, ..., n までを足し合わせるプロシージャを作成します．

```
> # ---------------------------------
  # 0, 1, 2, 3, ..., n までの足し算
  #
  # 引数： 正の整数（n）
  #
  # ---------------------------------
  mySum := proc( n::posint )

     # ローカル変数の定義
     local i, total;

     # 初期値の設定
     total := 0;

     # 0 から n までの加算
     for i from 1 to n do

        # 加算
        total := total + i;
```

```
    end do;

  end proc; # mySum
```

```
mySum := proc(n::posint)
    local i, total;
    total := 0; for i to n do total := total + i end do
end proc
```

(2.2.1.3.1)

n = 4 として，mySum を実行します．

```
> mySum(4);
```

10 **(2.2.1.3.2)**

n = 10 として，mySum を実行します．

```
> mySum(10);
```

55 **(2.2.1.3.3)**

（参考）sum コマンドを用いて実行します

数式入力 $\sum_{k=0}^{4} k$ をテキスト入力で実行します．

```
> sum('k', 'k'=0..4);
```

10 **(2.2.1.3.1.1)**

数式入力 $\sum_{k=0}^{10} k$ をテキスト入力で実行します．

```
> sum('k', 'k'=0..10);
```

55 **(2.2.1.3.1.2)**

sum を用いて，第 n 項を計算します．

```
> sum('k', 'k'=0..n);
```

$$\frac{(n+1)^2}{2} - \frac{n}{2} - \frac{1}{2}$$ **(2.2.1.3.1.3)**

確認のため，n=10 を代入してみます．

```
> eval((2.2.1.3.1.3), [ n=10 ]);
```

55 **(2.2.1.3.1.4)**

（参考）2つのインデックス操作

3×5 の行列 A を記号で定義します.

```
> A := Matrix(3, 5, symbol=a);
```

$$A := \begin{bmatrix} a_{1,1} & a_{1,2} & a_{1,3} & a_{1,4} & a_{1,5} \\ a_{2,1} & a_{2,2} & a_{2,3} & a_{2,4} & a_{2,5} \\ a_{3,1} & a_{3,2} & a_{3,3} & a_{3,4} & a_{3,5} \end{bmatrix} \quad \textbf{(2.2.2.1)}$$

各要素をインデックスといっしょにひとつずつ表示します．このケースでは i が行のインデックス，j が列のインデックスになります.

```
> for i from 1 to 3 do

    for j from 1 to 5 do

      # インデックス i, j,
      # 行列要素 a[i, j] の順番で表示
      print( i, j, A[i,j] );

    end do: # j

  end do: # i
```

$$1, 1, a_{1,1}$$
$$1, 2, a_{1,2}$$
$$1, 3, a_{1,3}$$
$$1, 4, a_{1,4}$$
$$1, 5, a_{1,5}$$
$$2, 1, a_{2,1}$$
$$2, 2, a_{2,2}$$
$$2, 3, a_{2,3}$$
$$2, 4, a_{2,4}$$
$$2, 5, a_{2,5}$$
$$3, 1, a_{3,1}$$
$$3, 2, a_{3,2}$$
$$3, 3, a_{3,3}$$
$$3, 4, a_{3,4}$$
$$3, 5, a_{3,5} \quad \textbf{(2.2.2.2)}$$

それは振動しますか？

２階常微分方程式の係数および初期条件を変化させることによって，解の変動を調べます．

初期化します．

```
> restart;
```

Maple による２階常微分方程式の解法と解のプロット

以下は，２階常微分方程式を定義するところから，必要な値や条件を定義して，解をプロットするところまでの処理になります．

① ２階常微分方程式を状態方程式に変換して定義します．

状態方程式とは， 行列表現に変換した１階の常微分方程式系になります．

（参考）「微分方程式に挑む！」の［２階常微分方程式の行列表現］を参照してください．

```
> ode := m*diff(x(t),t,t) + c*diff(x(t),t) + k*x(t) = 0;
```

$$ode := m\ddot{x}(t) + c\dot{x}(t) + k\,x(t) = 0 \qquad \textbf{(3.1.1)}$$

ode を行列表現にすると過程で以下の１階の常微分方程式が導かれます．

```
> deq1 := diff(x(t), t) = v(t);
```

$$deq1 := \dot{x}(t) = v(t) \qquad \textbf{(3.1.2)}$$

```
> deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);
```

$$deq2 := \dot{v}(t) = -\frac{k\,x(t)}{m} - \frac{c\,v(t)}{m} \qquad \textbf{(3.1.3)}$$

② 適当な係数 m，c，および k を定義します．

```
> params := [ m=1, c=1, k=6 ];
```

$$params := [m = 1, c = 1, k = 6] \qquad \textbf{(3.1.4)}$$

③ 定義した係数を deq1，deq2 に代入し，deq1p，deq2p をそれぞれ定義します．

```
> deq1p := eval( deq1, params );
  deq2p := eval( deq2, params );
```

$$deq1p := \dot{x}(t) = v(t)$$

$$deq2p := \dot{v}(t) = -6\,x(t) - v(t) \qquad \textbf{(3.1.5)}$$

④ 初期条件を定義します．

```
> ics := {x(0)=-1, D(x)(0)=0};
```

$$ics := \{x(0) = -1, D(x)(0) = 0\} \tag{3.1.6}$$

⑤ deq1p と deq2p を初期条件 ics として，以下のように連立して解きます．

```
> sol := dsolve( {deq1p, deq2p} union ics );
```

$$sol := \left\{ v(t) = \frac{12\, e^{-\frac{t}{2}} \sin\left(\frac{\sqrt{23}\, t}{2}\right)\sqrt{23}}{23}, x(t) = e^{-\frac{t}{2}}\left(-\frac{\sin\left(\frac{\sqrt{23}\, t}{2}\right)\sqrt{23}}{23} - \cos\left(\frac{\sqrt{23}\, t}{2}\right)\right)\right\} \tag{3.1.7}$$

⑥ 厳密解 sol から第２式の右辺を抽出して，eq に割り当てます．

```
> eq := rhs( sol[2] );
```

$$eq := e^{-\frac{t}{2}}\left(-\frac{\sin\left(\frac{\sqrt{23}\, t}{2}\right)\sqrt{23}}{23} - \cos\left(\frac{\sqrt{23}\, t}{2}\right)\right) \tag{3.1.8}$$

⑦ 求められた解をプロットします．

```
> plot( eq, t=0..10 );
```



「それは振動しますか？」プロシージャの作成手順

２階常微分方程式の解法をプロシージャ化

以上をプロシージャとしてまとめます．プロシージャ内で使用する変数をローカル変数として宣言します．

（注意） コメントは省略しています．

```
> mySim := proc()

      local deq1, deq2, params, deq1p, deq2p, ics, sol, eq;

      deq1 := diff(x(t), t) = v(t);
      deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

      params := [ m=1, c=1, k=6 ];
      deq1p := eval( deq1, params );
      deq2p := eval( deq2, params );

      ics := {x(0)=-1, D(x)(0)=0};
      sol := dsolve( {deq1p, deq2p} union ics );

      eq := rhs( sol[2] );
      plot( eq, t=0..10 );
```

```
end proc: # mySim
```

プロシージャの出力を非表示にするために，end proc の最後は，コロン（ : ）で閉じています．

mySim を実行します．

```
> mySim();
```



プロシージャのパラメータ化

係数 m，c，k，初期条件 x(0)，D(x)(0)，および時間区間 0..t の値を変化させて解をプロットするようなプロシージャに拡張します．すなわち，係数と初期条件をパラメータとして，解をプロットするプロシージャに変更します．わかりやすくするために一部のコメントをはずします．

初期化します．

```
> restart;
> mySim2 := proc(m, c, k, x0, dx0, et)
      local deq1, deq2, ics, sol, eq;
```

```
    deq1 := diff(x(t), t) = v(t);
    deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

    ics := {x(0)=x0, D(x)(0)=dx0};
    sol := dsolve( {deq1, deq2} union ics );

    eq := rhs( sol[2] );
    plot( eq, t=0..et );

end proc: # mySim2
```

mySim2 に係数 m, c, k, 初期条件 x0, dx0, 及び終了時間 et を順番に与えて実行します.

```
m   = 1
c   = 1
k   = 6
x0  = -1  ( x(0)=-1 )
dx0 = 0   ( D(x)(0)=0 )
et  = 10
```

```
> mySim2(1, 1, 6, -1, 0, 10);
```



パラメータをシーケンスとして定義します.

```
> sp := 1, 1, 6, -1, 0, 10;
```

$$sp := 1, 1, 6, -1, 0, 10 \qquad \textbf{(3.2.2.1)}$$

mySim2 に sp を引数として与えて実行します.

```
> mySim2(sp);
```



固有値の計算

ここで，プロシージャ内で係数行列 A の定義とその固有値を求める計算を追加します.

初期化します.

```
> restart;
> mySim3 := proc(m, c, k, x0, dx0, et)

      local A, evals, plt1, deq1, deq2, ics, sol, eq;

      A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
      evals := LinearAlgebra[Eigenvalues](A);

      deq1 := diff(x(t), t) = v(t);
      deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

      ics := {x(0)=x0, D(x)(0)=dx0};
      sol := dsolve( {deq1, deq2} union ics );
```

```
    eq := rhs( sol[2] );
    plt1 := plot( eq, t=0..et );

    return(plt1, evals);

end proc: # mySim3
```

パラメータを定義します.

```
> sp := 1, 1, 6, -1, 0, 10;
```

$$sp := 1, 1, 6, -1, 0, 10 \qquad \textbf{(3.2.3.1)}$$

mySim3 を sp で実行します.

```
> rslt := mySim3(sp);
```

$$rslt := PLOT(\ldots), \begin{bmatrix} -\frac{1}{2} + \frac{I\sqrt{23}}{2} \\ -\frac{1}{2} - \frac{I\sqrt{23}}{2} \end{bmatrix} \qquad \textbf{(3.2.3.2)}$$

プロットイメージを描画します.

```
> plots[display]( rslt[1] );
```



固有値の実部・虚部抽出

改めて，固有値のみを表示します．

```
> rslt[2];
```

$$\begin{bmatrix} -\frac{1}{2}+\frac{I\sqrt{23}}{2} \\ -\frac{1}{2}-\frac{I\sqrt{23}}{2} \end{bmatrix} \qquad \textbf{(3.2.4.1)}$$

固有値（特性方程式の根）から解の振る舞いが決定されます．そこで，m，c，k から固有値の性質（実部の符号，虚部のありなし）を判別し，解の振る舞いが振動的になるのか，そうならないのかを調べることにします．

ここで，処理の流れを見やすくするために，

```
# --------------------------------
```

を小さな処理のまとまりに対して追加します．

初期化します.

```
> restart;
> mySim4 := proc(m, c, k, x0, dx0, et)

    # --------------------------------
    local r1, i1, r2, i2, ri, A, evals, plt1,
          deq1, deq2, ics, sol, eq;

    # --------------------------------
    A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
    evals := LinearAlgebra[Eigenvalues](A);

    # --------------------------------
    r1 := Re(evals[1]);
    i1 := Im(evals[1]);
    r2 := Re(evals[2]);
    i2 := Im(evals[2]);
    ri := [r1, i1, r2, i2];

    # --------------------------------
    deq1 := diff(x(t), t) = v(t);
    deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

    # --------------------------------
    ics := {x(0)=x0, D(x)(0)=dx0};
    sol := dsolve( {deq1, deq2} union ics );

    # --------------------------------
    eq := rhs( sol[2] );
    plt1 := plot( eq, t=0..et );

    # --------------------------------
    return(plt1, evals, ri);

  end proc: # mySim4
```

パラメータを定義し，mySim4 を実行します.

```
> sp := 1, 1, 6, -1, 0, 10;
```

$$sp := 1, 1, 6, -1, 0, 10 \qquad \textbf{(3.2.4.2)}$$

```
> mySim4(sp);
```

$$PLOT(\ldots), \begin{bmatrix} -\frac{1}{2} + \frac{I\sqrt{23}}{2} \\ -\frac{1}{2} - \frac{I\sqrt{23}}{2} \end{bmatrix}, \left[-\frac{1}{2}, \frac{\sqrt{23}}{2}, -\frac{1}{2}, -\frac{\sqrt{23}}{2} \right] \qquad \textbf{(3.2.4.3)}$$

実部の符号と条件分岐

ここで，実部の符号から，その解の振る舞いを調べる処理（条件分岐）を追加します.

初期化します.

```
> restart;
> mySim5 := proc(m, c, k, x0, dx0, et)
```

```
      # --------------------------------
      local r1, i1, r2, i2, ri, A, evals, plt1,
            deq1, deq2, ics, sol, eq;

      # --------------------------------
      A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
      evals := LinearAlgebra[Eigenvalues](A);

      # --------------------------------
      r1 := Re(evals[1]);
      i1 := Im(evals[1]);
      r2 := Re(evals[2]);
      i2 := Im(evals[2]);
      ri := [r1, i1, r2, i2];

      # --------------------------------
      if r1 < 0 and r2 < 0 then

         # 収束します

      elif r1 > 0 or r2 > 0 then

         # 発散します

      end if;

      # --------------------------------
      deq1 := diff(x(t), t) = v(t);
      deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

      # --------------------------------
      ics := {x(0)=x0, D(x)(0)=dx0};
      sol := dsolve( {deq1, deq2} union ics );

      # --------------------------------
      eq := rhs( sol[2] );
      plt1 := plot( eq, t=0..et );

      # --------------------------------
      return(plt1, evals, ri);

   end proc: # mySim5
```

虚部の符号と条件分岐

続けて，虚部のありなしから，その解の振る舞いを調べる処理（条件分岐）を追加します．

```
> mySim6 := proc(m, c, k, x0, dx0, et)

      # --------------------------------
      local r1, i1, r2, i2, ri, A, evals, plt1,
            deq1, deq2, ics, sol, eq;

      # --------------------------------
      A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
      evals := LinearAlgebra[Eigenvalues](A);

      # --------------------------------
      r1 := Re(evals[1]);
      i1 := Im(evals[1]);
      r2 := Re(evals[2]);
      i2 := Im(evals[2]);
      ri := [r1, i1, r2, i2];
```

```
      # --------------------------------
      if r1 < 0 and r2 < 0 then

         if i1 = 0 or i2 = 0 then

            # 収束しますが，振動はしません.

         else

            # 振動しながら収束します.

         end if;


      elif r1 > 0 or r2 > 0 then

         if i1 = 0 or i2 = 0 then

            # 発散し，振動はしません.

         else

            # 振動しながら発散します.

         end if;

      end if;

      # --------------------------------
      deq1 := diff(x(t), t) = v(t);
      deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

      # --------------------------------
      ics := {x(0)=x0, D(x)(0)=dx0};
      sol := dsolve( {deq1, deq2} union ics );

      # --------------------------------
      eq := rhs( sol[2] );
      plt1 := plot( eq, t=0..et );

      # --------------------------------
      return(plt1, evals, ri);

   end proc: # mySim6
```

コメント部分を，文字列として出力します．evalf コマンドおよび evalb コマンドを追加します．return コマンドは，いったん無効にします（コメントにします）．

```
> mySim6 := proc(m, c, k, x0, dx0, et)

      # --------------------------------
      local r1, i1, r2, i2, ri, A, evals, plt1,
            deq1, deq2, ics, sol, eq;

      # --------------------------------
      A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
      evals := LinearAlgebra[Eigenvalues](A);

      # --------------------------------
      r1 := evalf( Re(evals[1]) );
      i1 := evalf( Im(evals[1]) );
      r2 := evalf( Re(evals[2]) );
      i2 := evalf( Im(evals[2]) );
      ri := [r1, i1, r2, i2];
```

```
      # --------------------------------
      if evalb( r1 < 0 ) and evalb( r2 < 0 ) then
         if i1 = 0 or i2 = 0 then
            print("収束しますが, 振動はしません. ");
         else
            print("振動しながら収束します. ");
         end if;

      elif evalb( r1 > 0 ) or evalb( r2 > 0 ) then
         if i1 = 0 or i2 = 0 then
            print("発散し, 振動はしません. ");
         else
            print("振動しながら発散します. ");
         end if;
      end if;

      # --------------------------------
      deq1 := diff(x(t), t) = v(t);
      deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

      # --------------------------------
      ics := {x(0)=x0, D(x)(0)=dx0};
      sol := dsolve( {deq1, deq2} union ics );

      # --------------------------------
      eq := rhs( sol[2] );
      plt1 := plot( eq, t=0..et );

      # --------------------------------
      # return(plt1, evals, ri);

   end proc: # mySim6
```

パラメータを定義し，mySim6 を実行します．

```
> sp := 1, 1, 6, -1, 0, 10;
```

$$sp := 1, 1, 6, -1, 0, 10 \qquad \textbf{(3.2.6.1)}$$

```
> mySim6(sp);
```

"振動しながら収束します. "



別のパラメータを定義し，mySim6 を実行します．

```
> sp :=  1,  5,  1, -1, 0, 10;
```

$$sp := 1, 5, 1, -1, 0, 10 \qquad \textbf{(3.2.6.2)}$$

```
> mySim6(sp);
```

"収束しますが，振動はしません．"



固有値のプロット

ここで，固有値を複素平面上にプロットします．

```
> mySim7 := proc(m, c, k, x0, dx0, et)

      # ---------------------------------
      local r1, i1, r2, i2, ri, A, evals, plt1, plt2, arrplt,
            deq1, deq2, ics, sol, eq;

      # ---------------------------------
      A := Matrix( 2,2, [0, 1, -k/m, -c/m]);
      evals := LinearAlgebra[Eigenvalues](A);

      # ---------------------------------
      r1 := evalf( Re(evals[1]) );
      i1 := evalf( Im(evals[1]) );
      r2 := evalf( Re(evals[2]) );
      i2 := evalf( Im(evals[2]) );
      ri := [r1, i1, r2, i2];

      # ---------------------------------
      if evalb( r1 < 0 ) and evalb( r2 < 0 ) then

         if i1 = 0 or i2 = 0 then

            print("収束しますが，振動はしません. ");
```

```
        else
           print("振動しながら収束します. ");
        end if;

     elif evalb( r1 > 0 ) or evalb( r2 > 0 ) then
        if i1 = 0 or i2 = 0 then
           print("発散し, 振動はしません. ");
        else
           print("振動しながら発散します. ");
        end if;
     end if;

     # ---------------------------------
     deq1 := diff(x(t), t) = v(t);
     deq2 := diff(v(t), t) = -(k/m)*x(t) - (c/m)*v(t);

     # ---------------------------------
     ics := {x(0)=x0, D(x)(0)=dx0};
     sol := dsolve( {deq1, deq2} union ics );

     # ---------------------------------
     eq := rhs( sol[2] );
     plt1 := plot( eq, t=0..et );

     plt2 := plots[complexplot]( convert(evals,list), style=point,
  symbolsize=24);

     arrplt := Array([plt2, plt1]);
     plots[display](arrplt);

     # ---------------------------------
     # return(plt1, evals, ri);

  end proc: # mySim7
```

パラメータを定義し，mySim7 を実行します.

```
> sp :=  1,  1,  1, -1, 0, 10;
```

$$sp := 1, 1, 1, -1, 0, 10 \qquad \textbf{(3.2.7.1)}$$

```
> mySim7(sp);
```

"振動しながら収束します. "


0.8
0.6
0.4
0.2
0
-0.2
-0.4
-0.6
-0.8
-0.7 -0.6 -0.5 -0.4 -0.3
0
2
4
6
8
10
t
-0.2
-0.4
-0.6
-0.8
-1

ステップ・バイ・ステップ式

はじめての Maple

応用編

Keywords & Key commands

やっぱり微積分

数列と和

```
seq()
sum()
plot()
infinity
' '
limit()
```

極限

```
limit()
plot()
limit( left )
limit( right )
plot()
```

微分法

```
diff()
simplify()
diff(), $
with( plots )
implicitplot()
sqrt()
plot( scaling )
expand()
```

連続微分

```
diff()
```

▼ 級数展開

```
series()
lhs()
rhs()
plot( view )
plot( numpoints )
convert( polynom )
plot()
seq()
with( plots )
display( insequence )
```

▼ 積分法

```
int()
simplify()
expand()
convert( tan )
convert( parfrac )
```

▼ 積分の応用

▼ 線の積分

```
plot()
diff()
sqrt()
int()
simplify()
->
evalf()
```

▼ 回転面の面積

```
->
int()
sqrt()
diff()
plot()
simplify()
eval()
plot( scaling )
```

▼ 2 重積分の計算

```
plot3d()
plot3d(axes
Int( Int() ) （大文字 I）
int( int() ) （小文字 i）
```

いわゆる線形代数

行列の演算

```
with( LinearAlgebra )
```

行列の定義

```
Matrix()
ZeroMatrix()
Matrix( symbol )
<>
,
|
IdentityMatrix()
DiagonalMatrix()
RandomMatrix()
map()
```

行列の演算

```
Matrix( symbol )
MatrixAdd()
-
.
^(-1)
MatrixInverse()
Determinant()
Rank()
MatrixNorm()
max()
```

行列の操作

```
with( LinearAlgebra )
Matrix( symbol )
Transpose(
^%T
[]
,
..
-1
```

ベクトル

```
with( LinearAlgebra )
```

ベクトルの定義

```
Vector()
ZeroVector()
Vector[row]()
Vector( symbol )
<>
|
,
RandomVector()
UnitVector()
```

ベクトルの操作

```
Vector()
Vector[row]()
[]
```

足し算

```
with( LinearAlgebra )
Vector( symbol )
VectorAdd()
```

▼ ベクトルの角度

```
Vector()
VectorAngle()
```

▼ 内積（スカラー積）と外積（ベクトル積）

```
with( Student[ VectorCalculus ] )
Vector()
PlotVector( color )
PlotVector( axes )
PlotVector( scaling )
DotProduct()
CrossProduct()
```

▼ ノルム（ベクトルの大きさ）

```
with( LinearAlgebra )
Vector()
Norm()
infinity
```

▼ 固有値と固有ベクトル

```
with( LinearAlgebra )
```

▼ 固有値の計算

```
Matrix()
Eigenvalues()
DiagonalMatrix()
Determinant()
solve()
```

▼ 固有ベクトルの計算

```
Eigenvectors()
```

▼ １次変換（線形写像）のプロット

```
with( Student[ LinearAlgebra ] )
IdentityMatrix()
LinearTransformPlot()
Matrix()
```

▼ 連立１次方程式の解法

```
with( LinearAlgebra )
eval()
solve()
GenerateMatrix( augmented )
LinearSolve()
MatrixInverse()
<>
|
true
.
ReducedRowEchelonForm()
LinearSystemPlot()
with( Student[ LinearAlgebra ] )
```

微分方程式に挑む！

数学モデルの作り方

常微分方程式の定義と解析解

少しだけ常微分方程式のこと

１階常微分方程式

（例１）　$\frac{\mathrm{d}}{\mathrm{d}x} y(x) = \left(1 - y(x)^2\right) \tan(x)$

```
diff()
dsolve()
subs()
rhs()
plots[animate]( plot )
```

（例２）　$x\left(\frac{\mathrm{d}}{\mathrm{d}x} y(x)\right) = y(x) + \sqrt{x^2 + y(x)^2}$

```
diff()
sqrt()
dsolve()
isolate()
eval()
rhs()
plots[animate]( plot )
```

▼ （例３） $\frac{\mathrm{d}}{\mathrm{d}x} y(x) + 2\, y(x) \tan(x) = \sin(x)$

```
diff()
tan()
sin()
dsolve()
expand()
subs()
rhs()
plots[animate]( plot )
```

▼ ２階常微分方程式の解

▼ １自由度粘性減衰振動系の解

```
diff()
exp()
eval()
expand()
solve()
dsolve()
```

▼ 指数関数の特性

```
exp()
```

▼ １） 異なる２つの実根を持つ場合

▼ a) ともに正の場合

```
eval()
plot()
+infinity
plots[complexplot]( style )
plots[complexplot]( symbolsize )
```

▼ b) 正と負の場合

```
eval()
plot()
+infinity
I
plots[complexplot]( style )
plots[complexplot]( symbolsize )
```

▼ c) ともに負の場合

```
eval()
plot()
+infinity
plots[complexplot]( style )
plots[complexplot]( symbolsize )
```

▼ 2) 異なる 2 つの虚根を持つ場合

▼ d) 実部が正の場合

```
eval()
evalc()
plot()
+infinity
plots[complexplot]( style )
plots[complexplot]( symbolsize )
```

▼ e) 実部が負の場合

```
eval()
evalc()
plot()
+infinity
plots[complexplot]( style )
plots[complexplot]( symbolsize )
```

▼ 少しだけ考察

▼ 2階常微分方程式の行列表現

```
diff()
eval()
isolate()
expand()
Vector()
Matrix()
map( diff )
LinearAlgebra[Eigenvalues]()
```

▼ 連立微分方程式の解法

```
dsolve( [ ], [ ] )
```

▼ 初期値問題

```
diff()
dsolve()
D()()
eval()
dsolve()
rhs()
plot()
plots[animate]( plot )
```

▼ ラプラス変換による初期値問題の解法

```
with( inttrans )
diff()
D()()
laplace()
subs()
eval()
isolate()
convert( parfrac )
invlaplace()
union
```

▼ 2 階常微分方程式の数値解法

```
diff()
eval()
D()()
dsolve( numeric )
plots[odeplot]()
plots[odeplot]( numpoints )
plots[odeplot]( frames )
```

▼プログラミング前夜

▼処理の再利用

▼プロシージャ化（処理を１行で記述可能な場合）

（基本構文）

プロシージャ名 := (変数) -> コマンド（変数を含む処理を記述）;

▼プロシージャ化（処理を複数行で記述する場合）

（基本構文）

```
プロシージャ名 := proc( 引数 )
   処理;
end proc;
```

▼グローバル変数とローカル変数

▼（例） グローバル変数とローカル変数の違い

```
proc()-end proc
local
```

▼引数の型・変数の型

（基本構文）

```
プロシージャ名 := proc( 引数::型 )::型
   local 変数::型
   処理;
end proc;
```

▼制御構文

▼条件分岐（if-then-else-end if）

（基本構文）

```
if 条件式１ then
   処理１;
elif 条件式２ then
   処理２;
else
   処理３;
end if;
```

▼数字の大小を判別する条件分岐

```
if-then-else-end if
evalb()
if-then-elif-then-else-end if
printf()
```

▼条件分岐を含むプロシージャの作成手順

▼（参考）引数を比較するプロシージャに変更

```
proc()-end proc
if-then-elif-then-else-end if
print()
```

▼繰り返し（for-do-end do）

（基本構文）

```
for インデックス名 from 開始数 by インクリメント数 to 終了数 do
```

処理（インデックス名を含む）;

```
end do;
```

▼ 足し算の実行

▼ 繰り返し文の前に

▼ 繰り返し文の利用

```
for-from-to-by-do-ebd do
printf()
for-from-to-do-ebd do
```

▼ プロシージャへの拡張

```
proc( posint )-local-end proc
for-from-to-do-ebd do
```

▼ （参考）sum コマンドを用いて実行します

```
sum()
' '
eval()
```

▼ （参考）２つのインデックス操作

```
Matrix( symbol )
for-from-to-do-ebd do
```

▼ それは振動しますか？

▼ Maple による２階常微分方程式の解法と解のプロット

```
diff()
eval()
D()()
```

```
dsolve()
union
rhs()
plot()
```

▼「それは振動しますか？」プロシージャの作成手順

▼ 2階常微分方程式の解法をプロシージャ化

---

proc()-local-end proc

▼ プロシージャのパラメータ化

---

proc(m, c, k, x0, dx0, et)-local-end proc

▼ 固有値の計算

---

proc(m, c, k, x0, dx0, et)-local-return-end proc

▼ 固有値の実部・虚部抽出

---

proc(m, c, k, x0, dx0, et)-local-return-end proc

▼ 実部の符号と条件分岐

```
proc( m, c, k, x0, dx0, et )-local-return-end proc
if-then-elif-then-else-end if
```

▼ 虚部の符号と条件分岐

```
proc( m, c, k, x0, dx0, et )-local-return-end proc
if-then-elif-then-else-end if
```

▼ 固有値のプロット

```
proc( m, c, k, x0, dx0, et )-local-return-end proc
if-then-elif-then-else-end if
```